

Extensionale Interpretation des Kategorischen Imperativs

Cornelius Diekmann

October 29, 2022

Contents

1	Schnelleinstieg Isabelle/HOL	1
1.1	Typen	1
1.2	Beweise	1
1.3	Mehr Typen	1
1.4	Funktionen	2
1.5	Mengen	2
2	Disclaimer	2
2.1	Über den Titel	3
3	Handlung	3
3.1	Interpretation: Gesinnungsethik vs. Verantwortungsethik	4
4	Gesetz	5
5	Kant's Kategorischer Imperativ	6
6	Beispiel Person	6
7	Maxime	7
7.1	Maxime in Sinne Kants?	7
7.2	Die Goldene Regel	8
7.3	Maximen Debugging	9
7.4	Beispiel	10
8	Schleier des Nichtwissens	11
9	Kategorischer Imperativ	13
9.1	Allgemeines Gesetz Ableiten	13
9.2	Implementierung Moralisch ein Allgemeines Gesetz Ableiten	14
9.3	Kategorischer Imperativ	15
9.4	Triviale Maximen die den Kategorischen Imperativ immer Erfüllen	17
9.5	Zusammenhang Goldene Regel	17

9.6	Maximen die den Kategorischen Imperativ immer Erfüllen	18
10	Utilitarismus	19
10.1	Goldene Regel und Utilitarismus im Einklang	19
11	Zahlenwelt Helper	20
12	Simulation	22
13	Gesetze	23
13.1	Case Law Absolut	23
13.2	Case Law Relativ	23
14	Beispiel: Zahlenwelt	23
14.1	Handlungen	24
14.2	Setup	26
14.3	Alice erzeugt 5 Wohlstand für sich.	26
14.4	Kleine Änderung in der Maxime	28
14.5	Maxime für Globales Optimum	29
14.6	Alice stiehlt 5	30
14.7	Schenken	31
14.8	Ungültige Maxime	31
15	Einkommensteuergesetzgebung	32
16	Beispiel: Steuern	35
16.1	Setup für Beispiele	36
16.2	Beispiel: Keiner Zahlt Steuern	36
16.3	Beispiel: Ich zahle 1 Steuer	36
16.4	Beispiel: Jeder zahle 1 Steuer	37
16.5	Beispiel: Vereinfachtes Deutsches Steuersystem	37
17	Vereinfachtes Deutsches Steuersystem vs. die Steuermaxime	38

1 Schnelleinstieg Isabelle/HOL

1.1 Typen

Typen werden per `::` annotiert. Beispielsweise sagt `3::nat`, dass `3` eine natürliche Zahl (*nat*) ist.

1.2 Beweise

Die besondere Fähigkeit im Beweisassistent Isabelle/HOL liegt darin, maschinengeprüfte Beweise zu machen.

Beispiel:

lemma $\langle 3 = 2 + 1 \rangle$

In der PDFversion wird der eigentliche Beweis ausgelassen. Aber keine Sorge, der Computer hat den Beweis überprüft. Würde der Beweis nicht gelten, würde das PDF garnicht compilieren.

Ich wurde schon für meine furchtbaren Beweise zitiert. Ist also ganz gut, dass wir nur Ergebnisse im PDF sehen und der eigentliche Beweis ausgelassen ist. Am besten kann man Beweise sowieso im Isabelle Editor anschauen und nicht im PDF.

1.3 Mehr Typen

Jeder Typ der mit einem einfachen Anführungszeichen anfängt ist ein polymorpher Typ. Beispiel: $'a$ oder $'\alpha$. So ein Typ ist praktisch ein generischer Typ, welcher durch jeden anderen Typen instanziiert werden kann.

Beispielsweise steht $'nat$ für einen beliebigen Typen, während nat der konkrete Typ der natürlichen Zahlen ist.

Wenn wir nun $3::'a$ schreiben handelt es sich nur um das generische Numeral 3. Das ist so generisch, dass z.B. noch nicht einmal die Plusoperation darauf definiert ist. Im Gegensatz dazu ist $3::nat$ die natürliche Zahl 3, mit allen wohlbekannten Rechenoperationen. Im Beweis obigen **lemma** $\langle 3 = 2 + 1 \rangle$ hat Isabelle die Typen automatisch inferiert.

1.4 Funktionen

Beispiel: Eine Funktionen welche eine natürliche Zahl nimmt und eine natürliche Zahl zurück gibt ($nat \Rightarrow nat$):

```
fun beispiefunktion ::  $\langle nat \Rightarrow nat \rangle$  where  
   $\langle beispiefunktion\ n = n + 10 \rangle$ 
```

Funktionsaufrufe funktionieren ohne Klammern.

lemma $\langle beispiefunktion\ 32 = 42 \rangle$

Funktionen sind gecurried. Hier ist eine Funktion welche 2 natürliche Zahlen nimmt und eine natürliche Zahl zurück gibt ($nat \Rightarrow nat \Rightarrow nat$):

```
fun addieren ::  $\langle nat \Rightarrow nat \Rightarrow nat \rangle$  where  
   $\langle addieren\ a\ b = a + b \rangle$ 
```

lemma $\langle addieren\ 32\ 10 = 42 \rangle$

Currying bedeutet auch, wenn wir *addieren* nur mit einem Argument aufrufen (welches eine natürliche Zahl *nat* sein muss), dass wir eine Funktion zurückbekommen, die noch das zweite Argument erwartet, bevor sie das Ergebnis zurückgeben kann.

Beispiel: $addieren\ 10::nat \Rightarrow nat$

Zufälligerweise ist $addieren\ 10$ equivalent zu *beispiefunktion*:

lemma $\langle addieren\ 10 = beispiefunktion \rangle$

Zusätzlich lassen sich Funktionen im Lambda Calculus darstellen. Beispiel:

lemma $\langle (\lambda n::nat. n+10) \ 3 = 13 \rangle$

lemma $\langle beispiefunktion = (\lambda n. n+10) \rangle$

1.5 Mengen

Mengen funktionieren wie normale mathematische Mengen.

Beispiel. Die Menge der geraden Zahlen:

lemma $\langle \{0,2,4,6,8,10,12\} \subseteq \{n::int. n \bmod 2 = 0\} \rangle$

2 Disclaimer

Ich habe

- kein Ahnung von Philosophie.
- keine Ahnung von Recht und Jura.
- und schon gar keine Ahnung von Strafrecht oder Steuerrecht.

Und in dieser Session werden ich all das zusammenwerfen.

Cheers!

2.1 Über den Titel

Der Titel lautet *Extensionale Interpretation des Kategorischen Imperativs*. Dabei sind die Wörter wie folgt zu verstehen

- *Extensional* bezieht sich hier auf den Fachbegriff der Logik <https://en.wikipedia.org/wiki/Extensionality>, welcher besagt, dass Objekte gleich sind, wenn sie die gleichen externen Eigenschaften aufweisen. Beispielsweise sind zwei Funktionen gleich, wenn sie für alle Eingaben die gleiche Ausgabe liefern: $(f = g) = (\forall x. f\ x = g\ x)$. Die interne (intensionale) Implementierung der Funktionen mag unterschiedlich sein, dennoch sind sie gleich. Dies ist die natürliche Gleichheit in HOL, welche uns erlaubt unser Modell bequem zu shallow-embedden. Meine extensionale Modellierung prägt diese Theorie stark. Beispielsweise sind Handlungen extensional modelliert, d.h nur die äußerlich messbaren Ergebnisse werden betrachtet. Dies widerspricht vermutlich stark Kants Vorstellung.

- *Interpretation* besagt, dass es sich hier um meine persönliche Interpretation handelt. Diese Theorie ist keine strenge Formalisierung der Literatur, sondern enthält sehr viele persönliche Meinungen.
- *Kategorischer Imperativ* bezieht sich auf Kants kategorischer Imperativ. Ziel dieser Theorie ist es, moralische Entscheidungen basierend auf Kants Idee zu machen.

3 Handlung

Beschreibt Handlungen als Änderung der Welt. Unabhängig von der handelnden Person. Wir beschreiben nur vergangene bzw. mögliche Handlungen und deren Auswirkung.

Eine Handlung ist reduziert auf deren Auswirkung. Intention oder Wollen ist nicht modelliert, da wir irgendwie die geistige Welt mit der physischen Welt verbinden müssen und wir daher nur messbare Tatsachen betrachten können.

Handlungen können Leute betreffen. Handlungen können aus Sicht Anderer wahrgenommen werden. Ich brauche nur Welt vorher und Welt nachher. So kann ich handelnde Person und beobachtende Person trennen.

datatype *'world handlung* = *Handlung* (*vorher*: *<'world>*) (*nachher*: *<'world>*)

Handlung als Funktion gewrapped. Diese abstrakte Art eine Handlung zu modelliert so ein bisschen die Absicht oder Intention.

datatype (*'person*, *'world*) *handlungF* = *HandlungF* *<'person ⇒ 'world ⇒ 'world>*

Von Außen können wir Funktionen nur extensional betrachten, d.h. Eingabe und Ausgabe anschauen. Die Absicht die sich in einer Funktion verstecken kann ist schwer zu erkennen. Dies deckt sich ganz gut damit, dass Isabelle standardmäßig Funktionen nicht printet. Eine (*'person*, *'world*) *handlungF* kann nicht geprinted werden!

fun *handeln* :: *<'person ⇒ 'world ⇒ ('person, 'world) handlungF ⇒ 'world handlung>* **where**
<handeln handelnde-person welt (HandlungF h) = Handlung welt (h handelnde-person welt)>

Beispiel, für eine Welt die nur aus einer Zahl besteht: Wenn die Zahl kleiner als 9000 ist erhöhe ich sie, ansonsten bleibt sie unverändert.

definition *<beispiel-handlungf ≡ HandlungF (λ- n. if n < 9000 then n+1 else n)>*

Da Funktionen nicht geprinted werden können, sieht *beispiel-handlungf* so aus: *HandlungF* -

3.1 Interpretation: Gesinnungsethik vs. Verantwortungsethik

Sei eine Ethik eine Funktion, welche einem beliebigen *'α* eine Bewertung Gut = *True*, Schlecht = *False* zuordnet.

- Eine Ethik hat demnach den Typ: *'α ⇒ bool*.

Laut <https://de.wikipedia.org/wiki/Gesinnungsethik> ist eine Gesinnungsethik "[...] eine der moralischen Theorien, die Handlungen nach der Handlungsabsicht [...] bewertet, und zwar ungeachtet der nach erfolgter Handlung eingetretenen Handlungsfolgen."

- Demnach ist eine Gesinnungsethik: $(\text{'person}, \text{'world}) \text{ handlung}F \Rightarrow \text{bool}$.

Nach <https://de.wikipedia.org/wiki/Verantwortungsethik> steht die Verantwortungsethik dazu im strikten Gegensatz, da die Verantwortungsethik "in der Bewertung des Handelns die Verantwortbarkeit der *tatsächlichen Ergebnisse* betont."

- Demnach ist eine Verantwortungsethik: $\text{'world handlung} \Rightarrow \text{bool}$.

Da *handeln* eine Handlungsabsicht $(\text{'person}, \text{'world}) \text{ handlung}F$ in eine konkrete Änderung der Welt 'world handlung überführt, können wie die beiden Ethiktypen miteinander in Verbindung setzen. Wir sagen, eine Gesinnungsethik und eine Verantwortungsethik sind konsistent, genau dann wenn für jede Handlungsabsicht, die Gesinnungsethik die Handlungsabsicht genau so bewertet, wie die Verantwortungsethik die Handlungsabsicht bewerten würde, wenn die die Handlungsabsicht in jeder möglichen Welt und als jede mögliche handelnde Person tatsächlich ausführt wird und die Folgen betrachtet werden:

definition *gesinnungsethik-verantwortungsethik-konsistent*

$:: \langle ((\text{'person}, \text{'world}) \text{ handlung}F \Rightarrow \text{bool}) \Rightarrow (\text{'world handlung} \Rightarrow \text{bool}) \Rightarrow \text{bool} \rangle$ **where**
 $\langle \text{gesinnungsethik-verantwortungsethik-konsistent gesinnungsethik verantwortungsethik} \equiv$
 $\forall \text{ handlungsabsicht.}$
 $\text{gesinnungsethik handlungsabsicht} \longleftrightarrow$
 $(\forall \text{ person welt. verantwortungsethik (handeln person welt handlungsabsicht)}) \rangle$

Ich habe kein Beispiel für eine Gesinnungsethik und eine Verantwortungsethik, die tatsächlich konsistent sind.

4 Gesetz

Definiert einen Datentyp um Gesetzestext zu modellieren.

datatype $\text{'a tatbestand} = \text{Tatbestand} \langle \text{'a} \rangle$

datatype $\text{'a rechtsfolge} = \text{Rechtsfolge} \langle \text{'a} \rangle$

datatype $(\text{'a}, \text{'b}) \text{ rechtsnorm} = \text{Rechtsnorm} \langle \text{'a tatbestand} \rangle \langle \text{'b rechtsfolge} \rangle$

datatype $\text{'p prg} = \text{Paragraph} \langle \text{'p} \rangle (\S)$

datatype $(\text{'p}, \text{'a}, \text{'b}) \text{ gesetz} = \text{Gesetz} \langle (\text{'p prg} \times (\text{'a}, \text{'b}) \text{ rechtsnorm}) \text{ set} \rangle$

Beispiel, von <https://de.wikipedia.org/wiki/Rechtsfolge>:

value $\langle \text{Gesetz} \{$

```

(§ "823 BGB",
 Rechtsnorm
  (Tatbestand "Wer vorsätzlich oder fahrlässig das Leben, den Körper, die Gesundheit, (...),
               das Eigentum oder (...) eines anderen widerrechtlich verletzt,")
  (Rechtsfolge "ist dem anderen zum Ersatz des daraus entstehenden Schadens verpflichtet.")
),
(§ "985 BGB",
 Rechtsnorm
  (Tatbestand "Der Eigentümer einer Sache kann von dem Besitzer")
  (Rechtsfolge "die Herausgabe der Sache verlangen")
),
(§ "303 StGB",
 Rechtsnorm
  (Tatbestand "Wer rechtswidrig eine fremde Sache beschädigt oder zerstört,")
  (Rechtsfolge "wird mit Freiheitsstrafe bis zu zwei Jahren oder mit Geldstrafe bestraft.")
)
}⟩

```

```

fun neuer-paragraph :: ⟨(nat, 'a, 'b) gesetz ⇒ nat prg⟩ where
  ⟨neuer-paragraph (Gesetz G) = § ((max-paragraph (fst ' G)) + 1)⟩

```

Fügt eine Rechtsnorm als neuen Paragraphen hinzu:

```

fun hinzufuegen :: ⟨('a, 'b) rechtsnorm ⇒ (nat, 'a, 'b) gesetz ⇒ (nat, 'a, 'b) gesetz⟩ where
  ⟨hinzufuegen rn (Gesetz G) =
    (if rn ∈ (snd ' G) then Gesetz G else Gesetz (insert (neuer-paragraph (Gesetz G), rn) G))⟩

```

Modelliert ob eine Handlung ausgeführt werden muss, darf, kann, nicht muss:

```

datatype sollensanordnung = Gebot | Verbot | Erlaubnis | Freistellung

```

Beispiel:

```

lemma ⟨hinzufuegen
  (Rechtsnorm (Tatbestand "tb2") (Rechtsfolge Verbot))
  (Gesetz { (§ 1, (Rechtsnorm (Tatbestand "tb1") (Rechtsfolge Erlaubnis))) }) =
Gesetz
  { (§ 2, Rechtsnorm (Tatbestand "tb2") (Rechtsfolge Verbot)),
    (§ 1, Rechtsnorm (Tatbestand "tb1") (Rechtsfolge Erlaubnis)) }⟩

```

5 Kant's Kategorischer Imperativ



Immanuel Kant

„Handle nur nach derjenigen *Maxime*, durch die du zugleich wollen kannst, dass sie ein allgemeines Gesetz werde.“

https://de.wikipedia.org/wiki/Kategorischer_Imperativ

Meine persönliche, etwas utilitaristische, Interpretation.

6 Beispiel Person

Eine Beispielbevölkerung.

datatype *person* = *Alice* | *Bob* | *Carol* | *Eve*

Unsere Bevölkerung ist sehr endlich:

lemma *UNIV-person*: $\langle UNIV = \{Alice, Bob, Carol, Eve\} \rangle$

Wir werden unterscheiden:

- *'person*: generischer Typ, erlaubt es jedes Modell einer Person und Bevölkerung zu haben.
- *person*: Unser minimaler Beispieltyp, bestehend aus *Alice*, *Bob*, ...

7 Maxime

Nach <https://de.wikipedia.org/wiki/Maxime> ist eine Maxime ein persönlicher Grundsatz des Wollens und Handelns. Nach Kant ist eine Maxime ein "subjektives Prinzip des Wollens".

Modell einer *Maxime*: Eine Maxime in diesem Modell beschreibt ob eine Handlung in einer gegebenen Welt gut ist.

Faktisch ist eine Maxime

- *'person*: die handelnde Person, i.e., *ich*.
- *'world handlung*: die zu betrachtende Handlung.
- *bool*: Das Ergebnis der Betrachtung. *True* = Gut; *False* = Schlecht.

Wir brauchen sowohl die *'world handlung* als auch die *'person* aus deren Sicht die Maxime definiert ist, da es einen großen Unterschied machen kann ob ich selber handel, ob ich Betroffener einer fremden Handlung bin, oder nur Außenstehender.

datatype (*'person*, *'world*) *maxime* = *Maxime* $\langle 'person \Rightarrow 'world\ handlung \Rightarrow bool \rangle$

Auswertung einer Maxime:

fun *okay* :: $\langle ('person, 'world)\ maxime \Rightarrow 'person \Rightarrow 'world\ handlung \Rightarrow bool \rangle$ **where**
 $\langle okay\ (Maxime\ m)\ p\ h = m\ p\ h \rangle$

Beispiel

definition *maxime-mir-ist-alles-recht* :: $\langle ('person, 'world)\ maxime \rangle$ **where**
 $\langle maxime-mir-ist-alles-recht \equiv Maxime\ (\lambda - .\ True) \rangle$

7.1 Maxime in Sinne Kants?

Kants kategorischer Imperativ ist eine deontologische Ethik, d.h., "Es wird eben nicht bewertet, was die Handlung bewirkt, sondern wie die Absicht beschaffen ist." https://de.wikipedia.org/wiki/Kategorischer_Imperativ.

Wenn wir Kants kategorischen Imperativ bauen wollen, dürfen wir also nicht die Folgen einer Handlung betrachten, sondern nur die Absicht dahinter. Doch unsere *Maxime* betrachtet eine *'world handlung*, also eine konkrete Handlung, die nur durch ihre Folgen gegeben ist. Die Maxime betrachtet keine Handlungsabsicht (*'person*, *'world*) *handlungF*.

Dies mag nun als Fehler in unserem Modell verstanden werden. Doch irgendwo müssen wir praktisch werden. Nur von Handlungsabsichten zu reden, ohne dass die beabsichtigten Folgen betrachtet werden ist mir einfach zu abstrakt und nicht greifbar.

Kants kategorischer Imperativ und die Goldene Regel grundverschieden: <https://web.archive.org/web/20220123174117/https://www.goethegymnasium-hildesheim.de/index.php/faecher/faecher/gesellschaftswissenschaften/philosophie> Dennoch, betrachten wir den kategorischen Imperativ als eine Verallgemeinerung der goldenen Regel.

7.2 Die Goldene Regel

Die Goldene Regel nach https://de.wikipedia.org/wiki/Goldene_Regel sagt:

„Behandle andere so, wie du von ihnen behandelt werden willst.“

„Was du nicht willst, dass man dir tu, das füg auch keinem andern zu.“

So wie wir behandelt werden wollen ist modelliert durch eine (*'person*, *'world*) *maxime*.

Die goldene Regel testet ob eine Handlung, bzw. Handlungsabsicht moralisch ist. Um eine Handlung gegen eine Maxime zu testen fragen wir uns:

- Was wenn jeder so handeln würde?
- Was wenn jeder nach dieser Maxime handeln würde?

Beispielsweise mag "stehlen" und "bestohlen werden" die gleiche Handlung sein, jedoch wird sie von Täter und Opfer grundverschieden wahrgenommen.

definition *bevoelkerung* :: $\langle 'person\ set \rangle$ **where** $\langle bevoelkerung \equiv UNIV \rangle$

definition *wenn-jeder-so-handelt*

:: $\langle 'world \Rightarrow ('person, 'world)\ handlungF \Rightarrow ('world\ handlung)\ set \rangle$

where

$\langle wenn-jeder-so-handelt\ welt\ handlungsabsicht \equiv$

$(\lambda handlende-person.\ handlen\ handlende-person\ welt\ handlungsabsicht)\ 'bevoelkerung \rangle$

fun *was-wenn-jeder-so-handelt-aus-sicht-von*

:: $\langle 'world \Rightarrow ('person, 'world)\ maxime \Rightarrow ('person, 'world)\ handlungF \Rightarrow 'person \Rightarrow bool \rangle$

where

$\langle was-wenn-jeder-so-handelt-aus-sicht-von\ welt\ m\ handlungsabsicht\ betroffene-person =$

$(\forall h \in wenn-jeder-so-handelt\ welt\ handlungsabsicht.\ okay\ m\ betroffene-person\ h) \rangle$

Für eine gegebene Welt und eine gegebene Maxime nennen wir eine Handlungsabsicht genau dann moralisch, wenn die Handlung auch die eigene Maxime erfüllt, wenn die Handlung von anderen durchgeführt würde.

definition *moralisch* ::

$\langle 'world \Rightarrow ('person, 'world)\ maxime \Rightarrow ('person, 'world)\ handlungF \Rightarrow bool \rangle$ **where**

$\langle moralisch\ welt\ handlungsabsicht\ maxime \equiv$

$\forall p \in bevoelkerung.\ was-wenn-jeder-so-handelt-aus-sicht-von\ welt\ handlungsabsicht\ maxime\ p \rangle$

Faktisch bedeutet diese Definition, wir bilden das Kreuzprodukt Bevölkerung x Bevölkerung, wobei jeder einmal als handelnde Person auftritt und einmal als betroffene Person.

lemma *moralisch-unfold*:

$\langle moralisch\ welt\ (Maxime\ m)\ handlungsabsicht \longleftrightarrow$

$(\forall p1 \in bevoelkerung.\ \forall p2 \in bevoelkerung.\ m\ p1\ (handeln\ p2\ welt\ handlungsabsicht)) \rangle$

lemma $\langle moralisch\ welt\ (Maxime\ m)\ handlungsabsicht \longleftrightarrow$

$(\forall (p1.p2) \in bevoelkerung \times bevoelkerung.\ m\ p1\ (handeln\ p2\ welt\ handlungsabsicht)) \rangle$

lemma *moralisch-simp*:

$\langle moralisch\ welt\ m\ handlungsabsicht \longleftrightarrow$

$(\forall p1\ p2.\ okay\ m\ p1\ (handeln\ p2\ welt\ handlungsabsicht)) \rangle$

Wir können die goldene Regel auch umformulieren, nicht als Imperativ, sondern als Beobachtung eines Wunschzustandes: Wenn eine Handlung für eine Person okay ist, dann muss sie auch Okay sein, wenn jemand anderes diese Handlung ausführt.

Formal: $m\ ich\ (handeln\ ich\ welt\ handlungsabsicht) \implies \forall p2.\ m\ ich\ (handeln\ p2\ welt\ handlungsabsicht)$

Genau dies können wir aus unserer Definition von *moralisch* ableiten:

lemma *goldene-regel*:

$\langle moralisch\ welt\ m\ handlungsabsicht \implies$

okay m ich (handeln ich welt handlungsabsicht) \implies
 $\forall p2. \text{okay } m \text{ ich (handeln } p2 \text{ welt handlungsabsicht)}$

Für das obige lemma brauchen wir die Annahme *m ich (handeln ich welt handlungsabsicht)* gar nicht. Wenn für eine gegebene *Maxime m* eine Handlungsabsicht moralisch ist, dann ist es auch okay, wenn ich von der Handlungsabsicht betroffen bin, egal wer sie ausführt.

corollary

$\langle \text{moralisch welt } m \text{ handlungsabsicht} \implies$
 $\forall p2. \text{okay } m \text{ ich (handeln } p2 \text{ welt handlungsabsicht)} \rangle$

Die umgekehrte Richtung gilt nicht, weil diese Formulierung nur die Handlungen betrachtet, die okay sind.

Hier schlägt das Programmiererherz höher: Wenn *'person* aufzählbar ist haben wir ausführbaren Code: *moralisch = moralisch-exhaust enum-class.enum* wobei *moralisch-exhaust* implementiert ist als *moralisch-exhaust bevoelk welt maxime handlungsabsicht \equiv case maxime of Maxime m \Rightarrow list-all* ($\lambda(p, x). m \text{ p (handeln } x \text{ welt handlungsabsicht)}$) (*List.product bevoelk bevoelk*).

7.3 Maximen Debugging

Der folgende Datentyp modelliert ein Beispiel in welcher Konstellation eine gegebene Maxime verletzt ist:

datatype *'person opfer = Opfer* $\langle 'person \rangle$
datatype *'person taeter = Taeter* $\langle 'person \rangle$
datatype (*'person, 'world*) *verletzte-maxime =*
VerletzteMaxime
 $\langle 'person \text{ opfer} \rangle$ — verletzt für; das Opfer
 $\langle 'person \text{ taeter} \rangle$ — handelnde Person; der Täter
 $\langle 'world \text{ handlung} \rangle$ — Die verletzende Handlung

Die folgende Funktion liefert alle Gegebenheiten welche eine Maxime verletzen:

fun *debug-maxime*
 $:: \langle ('world \Rightarrow 'printable-world) \Rightarrow 'world \Rightarrow$
 $(('person, 'world) \text{ maxime} \Rightarrow ('person, 'world) \text{ handlungF}$
 $\Rightarrow (('person, 'printable-world) \text{ verletzte-maxime}) \text{ set} \rangle$
where
 $\langle \text{debug-maxime print-world welt } m \text{ handlungsabsicht} =$
 $\{ \text{VerletzteMaxime}$
 $(\text{Opfer } p1) (\text{Taeter } p2)$
 $(\text{map-handlung print-world (handeln } p2 \text{ welt handlungsabsicht)}) \mid p1 \text{ } p2.$
 $\neg \text{okay } m \text{ } p1 \text{ (handeln } p2 \text{ welt handlungsabsicht)} \} \rangle$

Es gibt genau dann keine Beispiele für Verletzungen, wenn die Maxime erfüllt ist:

lemma $\langle \text{debug-maxime print-world welt maxime handlungsabsicht} = \{ \}$
 $\longleftrightarrow \text{moralisch welt maxime handlungsabsicht} \rangle$

7.4 Beispiel

Beispiel: Die Welt sei nur eine Zahl und die zu betrachtende Handlungsabsicht sei, dass wir diese Zahl erhöhen. Die Mir-ist-alles-Recht Maxime ist hier erfüllt:

lemma $\langle \text{moralisch}$
 $(42::\text{nat})$
 $\text{maxime-mir-ist-alles-recht}$
 $(\text{HandlungF } (\lambda(\text{person}::\text{person}) \text{ welt. welt} + 1)) \rangle$

Beispiel: Die Welt ist modelliert als eine Abbildung von Person auf Besitz. Die Maxime sagt, dass Leute immer mehr oder gleich viel wollen, aber nie etwas verlieren wollen. In einer Welt in der keiner etwas hat, erfüllt die Handlung jemanden 3 zu geben die Maxime.

lemma $\langle \text{moralisch}$
 $[\text{Alice} \mapsto (0::\text{nat}), \text{Bob} \mapsto 0, \text{Carol} \mapsto 0, \text{Eve} \mapsto 0]$
 $(\text{Maxime } (\lambda \text{person handlung.}$
 $(\text{the } ((\text{vorher handlung } \text{person})) \leq (\text{the } ((\text{nachher handlung } \text{person}))))$
 $(\text{HandlungF } (\lambda \text{person welt. welt}(\text{person} \mapsto 3)))) \rangle$

lemma $\langle \text{debug-maxime show-map}$
 $[\text{Alice} \mapsto (0::\text{nat}), \text{Bob} \mapsto 0, \text{Carol} \mapsto 0, \text{Eve} \mapsto 0]$
 $(\text{Maxime } (\lambda \text{person handlung.}$
 $(\text{the } ((\text{vorher handlung } \text{person})) \leq (\text{the } ((\text{nachher handlung } \text{person}))))$
 $(\text{HandlungF } (\lambda \text{person welt. welt}(\text{person} \mapsto 3))))$
 $= \{\}$

Wenn nun *Bob* allerdings bereits 4 hat, würde die obige Handlung ein Verlust für ihn bedeuten und die Maxime ist nicht erfüllt.

lemma $\langle \neg \text{moralisch}$
 $[\text{Alice} \mapsto (0::\text{nat}), \text{Bob} \mapsto 4, \text{Carol} \mapsto 0, \text{Eve} \mapsto 0]$
 $(\text{Maxime } (\lambda \text{person handlung.}$
 $(\text{the } ((\text{vorher handlung } \text{person})) \leq (\text{the } ((\text{nachher handlung } \text{person}))))$
 $(\text{HandlungF } (\lambda \text{person welt. welt}(\text{person} \mapsto 3)))) \rangle$

lemma $\langle \text{debug-maxime show-map}$
 $[\text{Alice} \mapsto (0::\text{nat}), \text{Bob} \mapsto 4, \text{Carol} \mapsto 0, \text{Eve} \mapsto 0]$
 $(\text{Maxime } (\lambda \text{person handlung.}$
 $(\text{the } ((\text{vorher handlung } \text{person})) \leq (\text{the } ((\text{nachher handlung } \text{person}))))$
 $(\text{HandlungF } (\lambda \text{person welt. welt}(\text{person} \mapsto 3))))$
 $= \{ \text{VerletzteMaxime } (\text{Opfer Bob}) (\text{Taeter Bob})$
 $(\text{Handlung } [(\text{Alice}, 0), (\text{Bob}, 4), (\text{Carol}, 0), (\text{Eve}, 0)]$
 $[(\text{Alice}, 0), (\text{Bob}, 3), (\text{Carol}, 0), (\text{Eve}, 0)]) \}$

8 Schleier des Nichtwissens

Rawls' Schleier des Nichtwissens https://de.wikipedia.org/wiki/Schleier_des_Nichtwissens ist ein fiktives Modell, » über die zukünftige Gesellschaftsordnung entscheiden können, aber selbst nicht wis-

sen, an welcher Stelle dieser Ordnung sie sich später befinden werden, also unter einem „Schleier des Nichtwissens“ stehen.« Quote wikipedia

Wir bedienen uns bei der Idee dieses Modells um gültige Handlungsabsichten und Maximen zu definieren. Handlungsabsichten und Maximen sind nur gültig, wenn darin keine Personen hardge-coded werden.

Beispielsweise ist folgende Handlungsabsicht ungültig: $\lambda ich\ welt. \text{ if } ich = Alice \text{ then } Do-A\ welt \text{ else } Do-B\ welt$

Handlungsabsichten und Maximen müssen immer generisch geschrieben werden, so dass die handelnden und betroffenen Personen niemals anhand ihres Namens ausgewählt werden.


unser Modell von Handlungsabsichten und Maximen stellt beispielsweise die handelnde Person als Parameter bereit. Folgendes ist also eine gültige Handlung: $\lambda ich\ welt. \text{ ModifiziereWelt } welt\ ich$

Auch ist es erlaubt, Personen in einer Handlungsabsicht oder Maxime nur anhand ihrer Eigenschaften in der Welt auszuwählen. Folgendes wäre eine wohlgeformte Handlung, wenn auch eine moralisch fragwürdige: $\lambda ich\ welt. \text{ enteignen } \{ \text{opfer. besitz } ich < \text{besitz opfer} \}$

Um diese Idee von wohlgeformten Handlungsabsichten und Maximen zu formalisieren bedienen wir uns der Idee des Schleiers des Nichtwissens. Wir sagen, dass Handlungsabsichten wohlgeformt sind, wenn die Handlungsabsicht gleich bleibt, wenn man sowohl die handelnde Person austauscht, als auch alle weltlichen Eigenschaften dieser Person. Anders ausgedrückt: Wohlgeformte Handlungsabsichten und Maximen sind solche, bei denen bei der Definition noch nicht feststeht, auf we sie später zutreffen.

Für jede Welt muss eine Welt-Personen Swap Funktion bereit gestellt werden, die alle Weltlichen Eigenschaften von 2 Personen vertauscht:

type-synonym $(\text{'person'}, \text{'world'})\ wp\text{-swap} = \langle \text{'person'} \Rightarrow \text{'person'} \Rightarrow \text{'world'} \Rightarrow \text{'world'} \rangle$



definition *wohlgeformte-handlungsabsicht*

$:: \langle (\text{'person'}, \text{'world'})\ wp\text{-swap} \Rightarrow \text{'world'} \Rightarrow (\text{'person'}, \text{'world'})\ handlungF \Rightarrow \text{bool} \rangle$

where

$\langle \text{wohlgeformte-handlungsabsicht } welt\text{-personen-swap } welt\ h \equiv$

$\forall p1\ p2. \text{ handeln } p1\ welt\ h =$

$\text{map-handlung } (welt\text{-personen-swap } p2\ p1) (\text{handeln } p2\ (welt\text{-personen-swap } p1\ p2\ welt)\ h) \rangle$

definition *wohlgeformte-handlungsabsicht-gegenbeispiel*

$:: \langle (\text{'person'}, \text{'world'})\ wp\text{-swap} \Rightarrow \text{'world'} \Rightarrow (\text{'person'}, \text{'world'})\ handlungF \Rightarrow \text{'person'} \Rightarrow \text{'person'} \Rightarrow \text{bool} \rangle$

where

$\langle \text{wohlgeformte-handlungsabsicht-gegenbeispiel } welt\text{-personen-swap } welt\ h\ taeter\ opfer \equiv$

$\text{handeln } taeter\ welt\ h \neq$

$\text{map-handlung } (welt\text{-personen-swap } opfer\ taeter) (\text{handeln } opfer\ (welt\text{-personen-swap } taeter\ opfer\ welt)\ h) \rangle$

lemma *wohlgeformte-handlungsabsicht-gegenbeispiel* $welt\text{-personen-swap } welt\ h\ p1\ p2 \implies$

$\neg \text{wohlgeformte-handlungsabsicht welt-personen-swap welt } h$

Nach der gleichen Argumentation müssen Maxime und Handlungsabsicht so generisch sein, dass sie in allen Welten zum gleichen Ergebnis kommen.

definition *maxime-und-handlungsabsicht-generalisieren*

$:: \langle ('person, 'world) \text{ maxime} \Rightarrow ('person, 'world) \text{ handlungF} \Rightarrow 'person \Rightarrow bool \rangle$

where

$\langle \text{maxime-und-handlungsabsicht-generalisieren } m \ h \ p =$
 $(\forall w1 \ w2. \text{ okay } m \ p \ (\text{handeln } p \ w1 \ h) \longleftrightarrow \text{ okay } m \ p \ (\text{handeln } p \ w2 \ h)) \rangle$

Die Maxime und $('person, 'world) \text{ wp-swap}$ müssen einige Eigenschaften erfüllen. Wir kürzen das ab mit wpsm: Welt Person Swap Maxime.

Die Person für die Maxime ausgewertet wird und swappen der Personen in der Welt muss equivalent sein:

definition *wpsm-kommutiert*

$:: \langle ('person, 'world) \text{ maxime} \Rightarrow ('person, 'world) \text{ wp-swap} \Rightarrow 'world \Rightarrow bool \rangle$

where

$\langle \text{wpsm-kommutiert } m \ \text{welt-personen-swap } \text{welt} \equiv$
 $\forall \ p1 \ p2 \ h.$
 $\text{ okay } m \ p2 \ (\text{Handlung } (\text{welt-personen-swap } p1 \ p2 \ \text{welt}) \ (h \ p1 \ (\text{welt-personen-swap } p1 \ p2 \ \text{welt})))$
 \longleftrightarrow
 $\text{ okay } m \ p1 \ (\text{Handlung } \text{welt} \ (\text{welt-personen-swap } p1 \ p2 \ (h \ p1 \ (\text{welt-personen-swap } p2 \ p1 \ \text{welt})))) \rangle$

lemma $\langle \text{wpsm-kommutiert } m \ \text{welt-personen-swap } \text{welt} =$

$(\forall \ p1 \ p2 \ h.$
 $\text{ okay } m \ p2 \ (\text{handeln } p1 \ (\text{welt-personen-swap } p1 \ p2 \ \text{welt}) \ (\text{HandlungF } h))$
 \longleftrightarrow
 $\text{ okay } m \ p1 \ (\text{handeln } p1 \ \text{welt} \ (\text{HandlungF } (\lambda p \ w. \text{ welt-personen-swap } p1 \ p2 \ (h \ p \ (\text{welt-personen-swap } p2 \ p1 \ w))))))$
 \rangle

Die Auswertung der Maxime für eine bestimmte Person muss unabhängig vom swappen von zwei unbeteiligten Personen sein.

definition *wpsm-unbeteiligt1*

$:: \langle ('person, 'world) \text{ maxime} \Rightarrow ('person, 'world) \text{ wp-swap} \Rightarrow 'world \Rightarrow bool \rangle$

where

$\langle \text{wpsm-unbeteiligt1 } m \ \text{welt-personen-swap } \text{welt} \equiv$
 $\forall \ p1 \ p2 \ pX \ \text{welt}.$
 $p1 \neq p2 \longrightarrow pX \neq p1 \longrightarrow pX \neq p2 \longrightarrow$
 $\text{ okay } m \ pX \ (\text{Handlung } (\text{welt-personen-swap } p2 \ p1 \ \text{welt}) \ \text{welt})$
 \longleftrightarrow
 $\text{ okay } m \ pX \ (\text{Handlung } \text{welt} \ \text{welt}) \rangle$

definition *wpsm-unbeteiligt2*

$:: \langle ('person, 'world) \text{ maxime} \Rightarrow ('person, 'world) \text{ wp-swap} \Rightarrow 'world \Rightarrow bool \rangle$

where

$\langle wpsm\text{-}unbeteiligt2\ m\ welt\text{-}personen\text{-}swap\ welt \equiv$
 $\forall\ p1\ p2\ pX\ h\ (welt\ ':\text{'world}).$
 $p1 \neq p2 \longrightarrow pX \neq p1 \longrightarrow pX \neq p2 \longrightarrow$
 $okay\ m\ pX\ (Handlung\ welt\ (welt\text{-}personen\text{-}swap\ p1\ p2\ (h\ p1\ welt\ ^)))$
 \longleftrightarrow
 $okay\ m\ pX\ (Handlung\ welt\ (h\ p1\ welt\ ^)) \rangle$

9 Kategorischer Imperativ

9.1 Allgemeines Gesetz Ableiten

Wir wollen implementieren:

„Handle nur nach derjenigen *Maxime*, durch die du zugleich wollen kannst, dass sie ein **allgemeines Gesetz** werde.“

Für eine gebene Welt haben wir schon eine Handlung nach einer *Maxime* untersucht: *moralisch*

Das Ergebnis sagt uns ob diese Handlung gut oder schlecht ist. Basierend darauf müssen wir nun ein allgemeines Gesetz ableiten.

Ich habe keine Ahnung wie das genau funktionieren soll, deswegen schreibe ich einfach nur in einer Typsignatur auf, was zu tun ist:

Gegeben:

- *'world handlung*: Die Handlung
- *sollensanordnung*: Das Ergebnis der moralischen Bewertung, ob die Handlung gut/schlecht.

Gesucht:

- *('a, 'b) rechtsnorm*: ein allgemeines Gesetz

type-synonym *('world, 'a, 'b) allgemeines-gesetz-ableiten =*
 $\langle 'world\ handlung \Rightarrow sollensanordnung \Rightarrow ('a, 'b)\ rechtsnorm \rangle$

Soviel vorweg: Nur aus einer von außen betrachteten Handlung und einer Entscheidung ob diese Handlung ausgeführt werden soll wird es schwer ein allgemeines Gesetz abzuleiten.

9.2 Implementierung Moralisch ein Allgemeines Gesetz Ableiten

Und nun werfen wir alles zusammen:

„Handle nur nach derjenigen *Maxime*, durch die du zugleich wollen kannst, dass sie ein **allgemeines Gesetz** werde.“

Eingabe:

- *'person*: handelnde Person
- *'world*: Die Welt in ihrem aktuellen Zustand
- *('person, 'world) handlungF*: Eine mögliche Handlung, über die wir entscheiden wollen ob wir sie ausführen sollten.
- *('person, 'world) maxime*: Persönliche Ethik.
- *('world, 'a, 'b) allgemeines-gesetz-ableiten*: wenn man keinen Plan hat wie man sowas implementiert, einfach als Eingabe annehmen.
- *(nat, 'a, 'b) gesetz*: Initiales allgemeines Gesetz (normalerweise am Anfang leer).

Ausgabe: *sollensanordnung*: Sollen wir die Handlung ausführen? *(nat, 'a, 'b) gesetz*: Soll das allgemeine Gesetz entsprechend angepasst werden?

definition *moarlich-gesetz-ableiten* ::

```

⟨ 'person ⇒
  'world ⇒
    ('person, 'world) maxime ⇒
    ('person, 'world) handlungF ⇒
    ('world, 'a, 'b) allgemeines-gesetz-ableiten ⇒
    (nat, 'a, 'b) gesetz
  ⇒ (sollensanordnung × (nat, 'a, 'b) gesetz)⟩

```

where

```

⟨ moarlich-gesetz-ableiten ich welt maxime handlungsabsicht gesetz-ableiten gesetz ≡
  let soll-handeln = if moralisch welt maxime handlungsabsicht
    then
      Erlaubnis
    else
      Verbot in
  (
    soll-handeln,
    hinzufuegen (gesetz-ableiten (handeln ich welt handlungsabsicht) soll-handeln) gesetz
  )⟩

```

Das ganze *moarlich-gesetz-ableiten* dient mehr dem Debugging, ...

9.3 Kategorischer Imperativ

Wir haben mit der goldenen Regel bereits definiert, wann für eine gegebene Welt und eine gegebene maxime, eine Handlungsabsicht moralisch ist:

- *moralisch::'world ⇒ ('person, 'world) maxime ⇒ ('person, 'world) handlungF ⇒ bool*

Effektiv testet die goldene Regel eine Handlungsabsicht.

Nach meinem Verständnis generalisiert Kant mit dem Kategorischen Imperativ diese Regel, indem die Maxime nicht mehr als gegeben angenommen wird, sondern die Maxime selbst getestet wird. Sei die Welt weiterhin gegeben, dass müsste der kategorische Imperativ folgende Typsignatur haben:

- $'world \Rightarrow ('person, 'world) \text{ maxime} \Rightarrow bool$

Eine Implementierung muss dann über alle möglichen Handlungsabsichten allquantifizieren.

Ich behaupte, der kategorischer Imperativ lässt sich wie folgt umformulieren:

- Handle nur nach derjenigen Maxime, durch die du zugleich wollen kannst, dass sie ein allgemeines Gesetz werde.
- Handle nur nach derjenigen Maxime, durch die du zugleich wollen kannst, dass sie jeder befolgt, im Sinne der goldenen Regel.
- Handle nur nach derjenigen Maxime, durch die du zugleich wollen kannst, dass sie (Handlung+Maxime) moralisch ist.
- Wenn es jemanden gibt der nach einer Maxime handeln will, dann muss diese Handlung nach der Maxime moralisch sein.
- Für jede Handlungsabsicht muss gelten: Wenn jemand in jeder Welt nach der Handlungsabsicht handeln würde, dann muss diese Handlung moralisch sein.

Daraus ergibt sich die finale Formalisierung:

Für alle möglichen (wohlgeformten) Handlungsabsichten muss gelten: Wenn es eine Person gibt für die diese Handlungsabsicht moralisch ist, dann muss diese Handlungsabsicht auch für alle moralisch (im Sinne der goldenen Regel) sein.

definition *kategorischer-imperativ*

$:: \langle ('person, 'world) \text{ wp-swap} \Rightarrow 'world \Rightarrow ('person, 'world) \text{ maxime} \Rightarrow bool \rangle$

where

$\langle \text{kategorischer-imperativ welt-personen-swap welt } m =$
 $(\forall h.$
 $\quad \text{wohlgeformte-handlungsabsicht welt-personen-swap welt } h \wedge$
 $\quad (\exists p. \text{ maxime-und-handlungsabsicht-generalisieren } m \ h \ p \wedge \text{ okay } m \ p \ (\text{handeln } p \ \text{welt } h))$
 $\quad \longrightarrow \text{ moralisch welt } m \ h) \rangle$

Dabei sind *wohlgeformte-handlungsabsicht* und *maxime-und-handlungsabsicht-generalisieren* ein technisch notwendiges Implementierungsdetail um nicht-wohlgeformte Handlungen oder Maximen auszuschließen.

Minimal andere Formulierung:

lemma

$\langle \text{kategorischer-imperativ welt-personen-swap welt } m \longleftrightarrow$
 $(\forall h.$
 $\quad (\exists p.$
 $\quad \text{wohlgeformte-handlungsabsicht welt-personen-swap welt } h \wedge$
 $\quad \text{maxime-und-handlungsabsicht-generalisieren } m \ h \ p \wedge$
 $\quad \text{okay } m \ p \ (\text{handeln } p \ \text{welt } h))$
 $\quad \longrightarrow \text{ moralisch welt } m \ h) \rangle$

Der Existenzquantor lässt sich auch in einen Allquantor umschreiben:

lemma

$\langle \text{kategorischer-imperativ welt-personen-swap welt } m \longleftrightarrow$
 $(\forall h \text{ ich.}$
 $\text{wohlgeformte-handlungsabsicht welt-personen-swap welt } h \wedge$
 $\text{maxime-und-handlungsabsicht-generalisieren } m \ h \text{ ich} \wedge$
 $\text{okay } m \text{ ich (handeln ich welt } h) \longrightarrow \text{moralisch welt } m \ h) \rangle$

Für jede Handlungsabsicht: wenn ich so handeln würde muss es auch okay sein, wenn zwei beliebige personen so handeln, wobei einer Täter und einer Opfer ist.

lemma kategorischer-imperativ-simp:

$\text{kategorischer-imperativ welt-personen-swap welt } m \longleftrightarrow$
 $(\forall h \ p1 \ p2 \text{ ich.}$
 $\text{wohlgeformte-handlungsabsicht welt-personen-swap welt } h \wedge$
 $\text{maxime-und-handlungsabsicht-generalisieren } m \ h \text{ ich} \wedge$
 $\text{okay } m \text{ ich (handeln ich welt } h)$
 $\longrightarrow \text{okay } m \ p1 \text{ (handeln } p2 \text{ welt } h))$

lemma kategorischer-imperativI:

$\langle (\bigwedge h \ p1 \ p2 \ p. \text{ wohlgeformte-handlungsabsicht welt-personen-swap welt } h \implies$
 $\text{maxime-und-handlungsabsicht-generalisieren } m \ h \ p \implies$
 $\text{okay } m \ p \text{ (handeln } p \text{ welt } h) \implies \text{okay } m \ p1 \text{ (handeln } p2 \text{ welt } h)$
 \rangle
 $\implies \text{kategorischer-imperativ welt-personen-swap welt } m \rangle$

9.4 Triviale Maximen die den Kategorischen Imperativ immer Erfüllen

Die Maxime die keine Handlung erlaubt (weil immer False) erfüllt den kategorischen Imperativ:

lemma $\langle \text{kategorischer-imperativ welt-personen-swap welt } (\text{Maxime } (\lambda \text{ich } h. \text{ False})) \rangle$

Allerdings kann mit so einer Maxime nie etwas moralisch sein.

lemma $\langle \neg \text{moralisch welt } (\text{Maxime } (\lambda \text{ich } h. \text{ False})) \ h \rangle$

Die Maxime die jede Handlung erlaubt (weil immer True) erfüllt den kategorischen Imperativ:

lemma $\langle \text{kategorischer-imperativ welt-personen-swap welt } (\text{Maxime } (\lambda \text{ich } h. \text{ True})) \rangle$

Allerdings ist mit so einer Maxime alles moralisch.

lemma $\langle \text{moralisch welt } (\text{Maxime } (\lambda \text{ich } h. \text{ True})) \ h \rangle$

9.5 Zusammenhang Goldene Regel

Mit der goldenen Regel konnten wir wie folgt moralische Entscheidungen treffen: $\llbracket \text{moralisch welt } m \text{ handlungsabsicht; okay m ich (handeln ich welt handlungsabsicht)} \rrbracket \implies \forall p2. \text{ okay m ich (handeln } p2 \text{ welt handlungsabsicht)}$

In Worten: Wenn eine Handlungsabsicht moralisch ist (nach goldener Regel) und es okay ist für mich diese Handlung auszuführen, denn ist es auch für mich okay, wenn jeder andere diese Handlung mit mir als Opfer ausführt.

Der kategorische Imperativ liftet dies eine Abstraktionsebene:

lemma $\langle \text{kategorischer-imperativ welt-personen-swap welt } m \implies$
 $\text{wohlgeformte-handlungsabsicht welt-personen-swap welt } h \implies$
 $\text{maxime-und-handlungsabsicht-generalisieren } m \text{ } h \text{ ich} \implies$
 $\text{okay m ich (handeln ich welt } h) \implies \text{moralisch welt } m \text{ } h \rangle$

In Worten: Wenn eine Maxime den kategorischen Imperativ erfüllt und es für eine beliebige (wohlgeformte) Handlung auszuführen für mich okay ist diese auszuführen, dann ist diese Handlung moralisch..

Für Beispiele wird es einfacher zu zeigen, dass eine Maxime nicht den kategorischen Imperativ erfüllt, wenn wir direkt ein Beispiel angeben.

definition $\text{kategorischer-imperativ-gegenbeispiel welt-personen-swap welt } m \text{ } h \text{ ich } p1 \text{ } p2 \equiv$
 $\text{wohlgeformte-handlungsabsicht welt-personen-swap welt } h \wedge$
 $\text{maxime-und-handlungsabsicht-generalisieren } m \text{ } h \text{ ich} \wedge$
 $\text{okay m ich (handeln ich welt } h) \wedge$
 $\neg \text{okay m } p1 \text{ (handeln } p2 \text{ welt } h)$

lemma $\text{kategorischer-imperativ-gegenbeispiel welt-personen-swap welt } m \text{ } h \text{ ich } p1 \text{ } p2 \implies$
 $\neg \text{kategorischer-imperativ welt-personen-swap welt } m$

9.6 Maximen die den Kategorischen Imperativ immer Erfüllen

Wenn eine Maxime jede Handlungsabsicht als moralisch bewertet, erfüllt diese Maxime den kategorischen Imperativ. Da diese Maxime jede Handlung erlaubt, ist es dennoch eine wohl ungeeignete Maxime.

lemma $\langle \forall h. \text{ moralisch welt maxime } h \implies \text{kategorischer-imperativ welt-personen-swap welt maxime} \rangle$

Eine Maxime die das ich und die Handlung ignoriert und etwas für alle fordert erfüllt den kategorischen Imperativ.

lemma $\text{blinde-maxime-katimp:}$
 $\langle \text{kategorischer-imperativ welt-personen-swap welt (Maxime } (\lambda \text{ich } h. \forall p. m \text{ } p)) \rangle$

Eine Maxime die das ich ignoriert und etwas für alle fordert erfüllt den kategorischen Imperativ.

theorem $\text{altruistische-maxime-katimp:}$

```

fixes  $P :: \langle 'person \Rightarrow 'world\ handlung \Rightarrow bool \rangle$ 
assumes  $kom: \langle wpsm-kommutiert\ (Maxime\ P)\ welt-personen-swap\ welt \rangle$ 
and  $unrel1: \langle wpsm-unbeteiligt1\ (Maxime\ P)\ welt-personen-swap\ welt \rangle$ 
and  $unrel2: \langle wpsm-unbeteiligt2\ (Maxime\ P)\ welt-personen-swap\ welt \rangle$ 
and  $welt-personen-swap-sym:$ 
 $\langle \forall p1\ p2\ welt. welt-personen-swap\ p1\ p2\ welt = welt-personen-swap\ p2\ p1\ welt \rangle$ 
shows  $\langle kategorischer-imperativ\ welt-personen-swap\ welt\ (Maxime\ (\lambda ich\ h. (\forall pX. P\ pX\ h))) \rangle$ 

```

Auch eine Maxime welche das ich ignoriert, also nur die Handlung global betrachtet, erfüllt den kategorischen Imperativ.

```

theorem globale-maxime-katimp:
fixes  $P :: \langle 'world\ handlung \Rightarrow bool \rangle$ 
assumes  $kom: \langle wpsm-kommutiert\ (Maxime\ (\lambda ich::'person. P))\ welt-personen-swap\ welt \rangle$ 
and  $unrel1: \langle wpsm-unbeteiligt1\ (Maxime\ (\lambda ich::'person. P))\ welt-personen-swap\ welt \rangle$ 
and  $unrel2: \langle wpsm-unbeteiligt2\ (Maxime\ (\lambda ich::'person. P))\ welt-personen-swap\ welt \rangle$ 
and  $welt-personen-swap-sym:$ 
 $\langle \forall p1\ p2\ welt. welt-personen-swap\ p1\ p2\ welt = welt-personen-swap\ p2\ p1\ welt \rangle$ 
shows  $\langle kategorischer-imperativ\ welt-personen-swap\ welt\ (Maxime\ (\lambda ich::'person. P)) \rangle$ 

```

10 Utilitarismus

Wir betrachten hier primär einen einfachen Handlungsutilitarismus. Frei nach Jeremy Bentham. Sehr frei. Also sehr viel persönliche Auslegung.

Eine Handlung ist genau dann moralisch richtig, wenn sie den aggregierten Gesamtnutzen, d.h. die Summe des Wohlergehens aller Betroffenen, maximiert wird.

type-synonym $'world\ glueck-messen = \langle 'world\ handlung \Rightarrow ereal \rangle$

Wir messen Glück im Typen *ereal*, also reelle Zahlen mit ∞ und $-\infty$, so dass auch "den höchsten Preis zahlen" modelliert werden kann.

lemma $\langle (\lambda h::ereal\ handlung. case\ h\ of\ Handlung\ vor\ nach \Rightarrow nach - vor)\ (Handlung\ 3\ 5) = 2 \rangle$

lemma $\langle (\lambda h::ereal\ handlung. case\ h\ of\ Handlung\ vor\ nach \Rightarrow nach - vor)\ (Handlung\ 3\ \infty) = \infty \rangle$

lemma $\langle (\lambda h::ereal\ handlung. case\ h\ of\ Handlung\ vor\ nach \Rightarrow nach - vor)\ (Handlung\ 3\ (-\infty)) = -\infty \rangle$

definition $moralisch-richtig :: \langle 'world\ glueck-messen \Rightarrow 'world\ handlung \Rightarrow bool \rangle$ **where**
 $\langle moralisch-richtig\ glueck-messen\ handlung \equiv (glueck-messen\ handlung) \geq 0 \rangle$

10.1 Goldene Regel und Utilitarismus im Einklang

In diese kleinen Intermezzo werden wir zeigen, wie sich die Gesinnungsethik der goldenen Regel in die Verantwortungsethik des Utilitarismus übersetzen lässt.

definition *goldene-regel-als-gesinnungsethik*
 $:: \langle ('person, 'world)\ maxime \Rightarrow ('person, 'world)\ handlungF \Rightarrow bool \rangle$

where

$\langle \text{goldene-regel-als-gesinnungsethik maxime handlungsabsicht} \equiv$
 $\forall \text{welt. moralisch welt maxime handlungsabsicht} \rangle$

definition *utilitarismus-als-verantwortungsethik*

$:: \langle 'world \text{ glueck-messen} \Rightarrow 'world \text{ handlung} \Rightarrow \text{bool} \rangle$

where

$\langle \text{utilitarismus-als-verantwortungsethik glueck-messen handlung} \equiv$
 $\text{moralisch-richtig glueck-messen handlung} \rangle$

Eine Maxime ist immer aus Sicht einer bestimmten Person definiert. Wir "neutralisieren" eine Maxime indem wir diese bestimmte Person entfernen und die Maxime so allgemeingültiger machen. Alle Personen müssen gleich behandelt werden Um die Maxime unabhängig von einer bestimmten Person zu machen, fordern wir einfach, dass die Maxime für aller Personen erfüllt sein muss.

fun *maximeNeutralisieren* $:: \langle ('person, 'world) \text{ maxime} \Rightarrow ('world \text{ handlung} \Rightarrow \text{bool}) \rangle$ **where**

$\langle \text{maximeNeutralisieren} (\text{Maxime } m) = (\lambda \text{welt. } \forall p::'person. m \text{ } p \text{ } \text{welt}) \rangle$

Nun übersetzen wir eine Maxime in die *'world glueck-messen* Funktion des Utilitarismus. Der Trick: eine verletzte Maxime wird als unendliches Leid übersetzt.

definition *maxime-als-nutzenkalkuel*

$:: \langle ('person, 'world) \text{ maxime} \Rightarrow 'world \text{ glueck-messen} \rangle$

where

$\langle \text{maxime-als-nutzenkalkuel maxime} \equiv$
 $(\lambda \text{welt. case } (\text{maximeNeutralisieren maxime}) \text{ welt}$
 $\text{ of True} \Rightarrow 1$
 $\text{ | False} \Rightarrow -\infty) \rangle$

Für diese Übersetzung können wir beweisen, dass die Gesinnungsethik der goldenen Regel und die utilitaristische Verantwortungsethik konsistent sind:

theorem *gesinnungsethik-verantwortungsethik-konsistent*

$(\text{goldene-regel-als-gesinnungsethik maxime})$
 $(\text{utilitarismus-als-verantwortungsethik } (\text{maxime-als-nutzenkalkuel maxime})) \rangle$

Diese Konsistenz gilt nicht im allgemeinen, sondern nur wenn Glück gemessen wird mit Hilfe der *maxime-als-nutzenkalkuel* Funktion. Der Trick dabei ist nicht, dass wir einer verletzten Maxime $-\infty$ Nutzen zuordnen, sondern der Trick besteht in *maximeNeutralisieren*, welche nicht erlaubt Glück aufzuaddieren und mit Leid zu verrechnen, sondern dank des Allquantors dafür sorgt, dass auch nur das kleinste Leid dazu führt, dass sofort *False* zurückgegeben wird.

Aber wenn wir ordentlich aufsummieren, jedoch einer verletzten Maxime $-\infty$ Nutzen zuordnen und zusätzlich annehmen, dass die Bevölkerung endlich ist, dann funktioniert das auch:

fun *maxime-als-summe-wohlergehen*

$:: \langle ('person, 'world) \text{ maxime} \Rightarrow 'world \text{ glueck-messen} \rangle$

where

$\langle \text{maxime-als-summe-wohlergehen} (\text{Maxime } m) =$
 $(\lambda \text{welt. } \sum p \in \text{bevoelkerung. (case } m \text{ } p \text{ } \text{welt}$

of True \Rightarrow 1
| False $\Rightarrow -\infty$)

theorem

fixes *maxime* :: $\langle ('person, 'world) maxime \rangle$

assumes $\langle finite (bevoelkerung:: 'person set) \rangle$

shows

$\langle gesinnungsethik-verantwortungsethik-konsistent$

$(goldene-regel-als-gesinnungsethik maxime)$

$(utilitarismus-als-verantwortungsethik (maxime-als-summe-wohlergehen maxime)) \rangle$

11 Zahlenwelt Helper

Wir werden Beispiele betrachten, in denen wir Welten modellieren, in denen jeder Person eine Zahl zugewiesen wird: $person \Rightarrow int$. Diese Zahl kann zum Beispiel der Besitz oder Wohlstand einer Person sein, oder das Einkommen. Wobei Gesamtbesitz und Einkommen über einen kurzen Zeitraum recht unterschiedliche Sachen modellieren.

Hier sind einige Hilfsfunktionen um mit $person \Rightarrow int$ allgemein zu arbeiten.

Default: Standardmäßig hat jede Person 0:

definition *DEFAULT* :: $\langle person \Rightarrow int \rangle$ **where**

$\langle DEFAULT \equiv \lambda p. 0 \rangle$

Beispiel:

lemma $\langle (DEFAULT(Alice:=8, Bob:=3, Eve:= 5)) Bob = 3 \rangle$

Beispiel mit fancy Syntax:

lemma $\langle \clubsuit[Alice:=8, Bob:=3, Eve:= 5] Bob = 3 \rangle$

lemma $\langle show-fun \clubsuit[Alice := 4, Carol := 4] = [(Alice, 4), (Bob, 0), (Carol, 4), (Eve, 0)] \rangle$

lemma $\langle show-num-fun \clubsuit[Alice := 4, Carol := 4] = [(Alice, 4), (Carol, 4)] \rangle$

abbreviation *num-fun-add-syntax* (- '(- += -')) **where**

$\langle f(p += n) \equiv (f(p := (f p) + n)) \rangle$

abbreviation *num-fun-minus-syntax* (- '(- -= -')) **where**

$\langle f(p -= n) \equiv (f(p := (f p) - n)) \rangle$

lemma $\langle (\clubsuit[Alice:=8, Bob:=3, Eve:= 5])(Bob += 4) Bob = 7 \rangle$

lemma $\langle (\clubsuit[Alice:=8, Bob:=3, Eve:= 5])(Bob -= 4) Bob = -1 \rangle$

lemma *fixes n* :: *<int>* **shows** $\langle f(p \text{ += } n)(p \text{ -= } n) = f \rangle$

Diskriminierungsfrei eine *'person* eindeutig anhand Ihres Besitzes auswählen:

definition *opfer-eindeutig-nach-besitz-auswaehlen*

```
:: <int  $\Rightarrow$  ('person  $\Rightarrow$  int)  $\Rightarrow$  'person list  $\Rightarrow$  'person option> where
<opfer-eindeutig-nach-besitz-auswaehlen b besitz ps =
  (case filter ( $\lambda p$ . besitz p = b) ps
    of [opfer]  $\Rightarrow$  Some opfer
      | -  $\Rightarrow$  None)>
```

definition *the-single-elem* :: *<'a set \Rightarrow 'a option>* **where**

```
<the-single-elem s  $\equiv$  if card s = 1 then Some (Set.the-elem s) else None>
```

thm *is-singleton-the-elem* [*symmetric*]

lemma $A = \{\text{the-elem } A\} \longleftrightarrow \text{is-singleton } A$

lemma *opfer-nach-besitz-induct-step-set-simp*: *besitz a \neq opfer-nach-besitz \implies*

```
{p. (p = a  $\vee$  p  $\in$  set ps)  $\wedge$  besitz p = opfer-nach-besitz} =
{p  $\in$  set ps. besitz p = opfer-nach-besitz}
```

lemma *opfer-eindeutig-nach-besitz-auswaehlen-the-single-elem*:

```
<distinct ps  $\implies$ 
  opfer-eindeutig-nach-besitz-auswaehlen opfer-nach-besitz besitz ps =
  the-single-elem {p  $\in$  set ps. besitz p = opfer-nach-besitz}>
```

lemma *opfer-eindeutig-nach-besitz-auswaehlen-the-single-elem-enumall*:

```
<opfer-eindeutig-nach-besitz-auswaehlen opfer-nach-besitz besitz enum-class.enum =
  the-single-elem {p. besitz p = opfer-nach-besitz}>
```

12 Simulation

Gegeben eine handelnde Person und eine Maxime, wir wollen simulieren was für ein allgemeines Gesetz abgeleitet werden könnte.

datatype (*'person*, *'world*, *'a*, *'b*) *simulation-constants* = *SimConsts*

```
<'person> — handelnde Person
<('person, 'world) maxime>
<('world, 'a, 'b) allgemeines-gesetz-ableiten>
```

...

... Die Funktion *simulateOne* nimmt eine Konfiguration (*'person*, *'world*, *'a*, *'b*) *simulation-constants*, eine Anzahl an Iterationen die durchgeführt werden sollen, eine Handlung, eine Initialwelt, ein Initialgesetz, und gibt das daraus resultierende Gesetz nach so vielen Iterationen zurück.

Beispiel: Wir nehmen die mir-ist-alles-egal Maxime. Wir leiten ein allgemeines Gesetz ab indem wir einfach nur die Handlung wörtlich ins Gesetz übernehmen. Wir machen *10::'a* Iterationen. Die Welt ist nur eine Zahl und die initiale Welt sei *32::'a*. Die Handlung ist es diese Zahl um Eins zu erhöhen, Das Ergebnis der Simulation ist dann, dass wir einfach von *32::'a* bis *42::'a* zählen.

lemma $\langle \text{simulateOne}$
 (*SimConsts* ()) (*Maxime* ($\lambda - . \text{True}$)) ($\lambda h s. \text{Rechtsnorm (Tatbestand } h) (\text{Rechtsfolge "count"})$)
 10 (*HandlungF* ($\lambda p n. \text{Suc } n$))
 32
 (*Gesetz* {})) =
Gesetz
 { (§ 10, *Rechtsnorm (Tatbestand (Handlung 41 42)) (Rechtsfolge "count")*),
 (§ 9, *Rechtsnorm (Tatbestand (Handlung 40 41)) (Rechtsfolge "count")*),
 (§ 8, *Rechtsnorm (Tatbestand (Handlung 39 40)) (Rechtsfolge "count")*),
 (§ 7, *Rechtsnorm (Tatbestand (Handlung 38 39)) (Rechtsfolge "count")*),
 (§ 6, *Rechtsnorm (Tatbestand (Handlung 37 38)) (Rechtsfolge "count")*),
 (§ 5, *Rechtsnorm (Tatbestand (Handlung 36 37)) (Rechtsfolge "count")*),
 (§ 4, *Rechtsnorm (Tatbestand (Handlung 35 36)) (Rechtsfolge "count")*),
 (§ 3, *Rechtsnorm (Tatbestand (Handlung 34 35)) (Rechtsfolge "count")*),
 (§ 2, *Rechtsnorm (Tatbestand (Handlung 33 34)) (Rechtsfolge "count")*),
 (§ 1, *Rechtsnorm (Tatbestand (Handlung 32 33)) (Rechtsfolge "count")*)} \rangle

Eine Iteration der Simulation liefert genau einen Paragraphen im Gesetz:

lemma $\langle \exists tb \text{ rf.}$
simulateOne
 (*SimConsts person maxime gesetz-ableiten*)
 1 *handlungsabsicht*
initialwelt
 (*Gesetz* {}))
 = *Gesetz* { (§ 1, *Rechtsnorm (Tatbestand tb) (Rechtsfolge rf)*) } \rangle

13 Gesetze

Wir implementieren Strategien um (*'world, 'a, 'b*) *allgemeines-gesetz-ableiten* zu implementieren.

13.1 Case Law Absolut

Gesetz beschreibt: wenn (vorher, nachher) dann Erlaubt/Verboten, wobei vorher/nachher die Welt beschreiben. Paragraphen sind einfache natürliche Zahlen.

type-synonym *'world case-law* = $\langle (\text{nat}, ('world \times 'world), \text{sollensanordnung}) \text{ gesetz} \rangle$

Überträgt einen Tatbestand wörtlich ins Gesetz. Nicht sehr allgemein.

definition *case-law-ableiten-absolut*
 $:: \langle ('world, ('world \times 'world), \text{sollensanordnung}) \text{ allgemeines-gesetz-ableiten} \rangle$
where
 $\langle \text{case-law-ableiten-absolut handlung sollensanordnung} =$
Rechtsnorm
 (*Tatbestand (vorher handlung, nachher handlung)*)
 (*Rechtsfolge sollensanordnung*) \rangle

definition *printable-case-law-ableiten-absolut*

```
:: ⟨('world ⇒ 'printable-world) ⇒
  ('world, ('printable-world × 'printable-world), sollensanordnung) allgemeines-gesetz-ableiten⟩
where
⟨printable-case-law-ableiten-absolut print-world h ≡
  case-law-ableiten-absolut (map-handlung print-world h)⟩
```

13.2 Case Law Relativ

Case Law etwas besser, wir zeigen nur die Änderungen der Welt.

fun *case-law-ableiten-relativ*

```
:: ⟨('world handlung ⇒ (('person, 'etwas) aenderung) list)
  ⇒ ('world, (('person, 'etwas) aenderung) list, sollensanordnung)
  allgemeines-gesetz-ableiten⟩
```

where

```
⟨case-law-ableiten-relativ delta handlung erlaubt =
  Rechtsnorm (Tatbestand (delta handlung)) (Rechtsfolge erlaubt)⟩
```

14 Beispiel: Zahlenwelt

Wir nehmen an, die Welt lässt sich durch eine Zahl darstellen, die den Besitz einer Person modelliert. Der Besitz ist als ganze Zahl *int* modelliert und kann auch beliebig negativ werden.

datatype *zahlenwelt* = *Zahlenwelt*

```
⟨person ⇒ int — besitz: Besitz jeder Person.⟩
```

fun *gesamtbesitz* :: ⟨*zahlenwelt* ⇒ *int*⟩ **where**

```
⟨gesamtbesitz (Zahlenwelt besitz) = sum-list (map besitz Enum.enum)⟩
```

lemma ⟨*gesamtbesitz* (Zahlenwelt besitz) = $(\sum p \leftarrow [Alice, Bob, Carol, Eve]. \text{besitz } p)$ ⟩

lemma ⟨*gesamtbesitz* (Zahlenwelt ♣[Alice := 4, Carol := 8]) = 12⟩

lemma ⟨*gesamtbesitz* (Zahlenwelt ♣[Alice := 4, Carol := 4]) = 8⟩

fun *meins* :: ⟨*person* ⇒ *zahlenwelt* ⇒ *int*⟩ **where**

```
⟨meins p (Zahlenwelt besitz) = besitz p⟩
```

lemma ⟨*meins* Carol (Zahlenwelt ♣[Alice := 8, Carol := 4]) = 4⟩

fun *zahlenwelt-personen-swap* :: ⟨*person* ⇒ *person* ⇒ *zahlenwelt* ⇒ *zahlenwelt*⟩ **where**

```
⟨zahlenwelt-personen-swap p1 p2 (Zahlenwelt besitz) = Zahlenwelt (swap p1 p2 besitz)⟩
```

lemma ⟨*zahlenwelt-personen-swap* Alice Carol (Zahlenwelt ♣[Alice := 4, Bob := 6, Carol := 8])
= (Zahlenwelt ♣[Alice := 8, Bob := 6, Carol := 4])⟩

14.1 Handlungen

Die folgende Handlung erschafft neuen Besitz aus dem Nichts:

```
fun erschaffen :: <nat ⇒ person ⇒ zahlenwelt ⇒ zahlenwelt> where
  <erschaffen i p (Zahlenwelt besitz) = Zahlenwelt (besitz(p += int i))>
lemma <wohlgeformte-handlungsabsicht zahlenwelt-personen-swap welt (HandlungF (erschaffen n))>
```

```
fun stehlen :: <int ⇒ person ⇒ person ⇒ zahlenwelt ⇒ zahlenwelt> where
  <stehlen beute opfer dieb (Zahlenwelt besitz) =
    Zahlenwelt (besitz(opfer -= beute)(dieb += beute))>
```

Die Handlung *stehlen* diskriminiert und ist damit nicht wohlgeformt:

```
lemma wohlgeformte-handlungsabsicht-gegenbeispiel zahlenwelt-personen-swap
  (Zahlenwelt (λx. 0)) (HandlungF (stehlen 5 Bob))
  Alice Bob
```

Wir versuchen, das Opfer nach Besitz auszuwählen, nicht nach Namen. Nach unserer Definition ist der Besitz ein Merkmal, nach dem man diskriminieren darf. Man darf nur nicht nach Eigenschaften der *person* diskriminieren, sondern nur nach Eigenschaften der *zahlenwelt*.

```
fun opfer-nach-besitz-auswaehlen :: <int ⇒ ('person ⇒ int) ⇒ 'person list ⇒ 'person option> where
  <opfer-nach-besitz-auswaehlen - - [] = None>
  | <opfer-nach-besitz-auswaehlen b besitz (p#ps) =
    (if besitz p = b then Some p else opfer-nach-besitz-auswaehlen b besitz ps)>
```

```
fun stehlen2 :: <int ⇒ int ⇒ person ⇒ zahlenwelt ⇒ zahlenwelt> where
  <stehlen2 beute opfer-nach-besitz dieb (Zahlenwelt besitz) =
    (case opfer-nach-besitz-auswaehlen opfer-nach-besitz besitz Enum.enum
      of None ⇒ (Zahlenwelt besitz)
       | Some opfer ⇒ Zahlenwelt (besitz(opfer -= beute)(dieb += beute))
    )>
```

Leider ist diese Funktion auch diskriminierend: Wenn es mehrere potenzielle Opfer mit dem gleichen Besitz gibt, dann bestimmt die Reihenfolge in *enum-class.enum* wer bestohlen wird. Diese Reihenfolge ist wieder eine Eigenschaft von *person* und nicht *zahlenwelt*.

```
lemma<handeln Alice (Zahlenwelt ♠[Alice := 10, Bob := 10, Carol := -3])
  (HandlungF (stehlen2 5 10))
= Handlung (Zahlenwelt ♠[Alice := 10, Bob := 10, Carol := -3])
  (Zahlenwelt ♠[Alice := 10, Bob := 10, Carol := -3])>
lemma<handeln Bob (Zahlenwelt ♠[Alice := 10, Bob := 10, Carol := -3])
  (HandlungF (stehlen2 5 10))
= Handlung (Zahlenwelt ♠[Alice := 10, Bob := 10, Carol := -3])
  (Zahlenwelt ♠[Alice := 5, Bob := 15, Carol := -3])>
lemma <wohlgeformte-handlungsabsicht-gegenbeispiel
  zahlenwelt-personen-swap
  (Zahlenwelt ♠[Alice := 10, Bob := 10, Carol := -3]) (HandlungF (stehlen2 5 10))
  Alice Bob>
```

Wenn wir das Opfer allerdings eindeutig auswählen, ist die Handlung wohlgeformt. Allerdings wird niemand bestohlen, wenn das Opfer nicht eindeutig ist.

```
fun stehlen4 :: ⟨int ⇒ int ⇒ person ⇒ zahlenwelt ⇒ zahlenwelt⟩ where
  ⟨stehlen4 beute opfer-nach-besitz dieb (Zahlenwelt besitz) =
    (case opfer-eindeutig-nach-besitz-auswaehlen opfer-nach-besitz besitz Enum.enum
      of None ⇒ (Zahlenwelt besitz)
       | Some opfer ⇒ Zahlenwelt (besitz(opfer -= beute)(dieb += beute))
    )⟩
```

lemma wohlgeformte-handlungsabsicht-stehlen4:

⟨wohlgeformte-handlungsabsicht zahlenwelt-personen-swap welt (HandlungF (stehlen4 n p))⟩

```
fun schenken :: ⟨int ⇒ person ⇒ person ⇒ zahlenwelt ⇒ zahlenwelt⟩ where
```

```
  ⟨schenken betrag empfaenger schenker (Zahlenwelt besitz) =
    Zahlenwelt (besitz(schenker -= betrag)(empfaenger += betrag))⟩
```

Da wir ganze Zahlen verwenden und der Besitz auch beliebig negativ werden kann, ist Stehlen äquivalent dazu einen negativen Betrag zu verschenken:

lemma stehlen-ist-schenken: ⟨stehlen i = schenken (−i)⟩

Das Modell ist nicht ganz perfekt, Aber passt schon um damit zu spielen.

Reset versetzt die Welt wieder in den Ausgangszustand. Eine sehr destruktive Handlung.

```
fun reset :: ⟨person ⇒ zahlenwelt ⇒ zahlenwelt⟩ where
  ⟨reset ich (Zahlenwelt besitz) = Zahlenwelt (λ -. 0)⟩
```

Der *reset* ist im moralischen Sinne vermutlich keine gute Handlung, dennoch ist es eine wohlgeformte Handlung, welche wir betrachten können:

lemma ⟨wohlgeformte-handlungsabsicht zahlenwelt-personen-swap welt (HandlungF reset)⟩

14.2 Setup

Alice hat Besitz, Bob ist reicher, Carol hat Schulden.

definition ⟨initialwelt ≡ Zahlenwelt ♣[Alice := 5, Bob := 10, Carol := −3]⟩

Wir nehmen an unsere handelnde Person ist Alice.

```
definition ⟨beispiel-case-law-absolut maxime handlungsabsicht ≡
  simulateOne
    (SimConsts
      Alice
      maxime
      (printable-case-law-ableiten-absolut show-zahlenwelt))
  5 handlungsabsicht initialwelt (Gesetz {})⟩
```

definition $\langle \text{beispiel-case-law-relativ maxime handlungsabsicht} \equiv$
 simulateOne
 $(\text{Sim Consts}$
 Alice
 maxime
 $(\text{case-law-ableiten-relativ delta-zahlenwelt}))$
 $10 \text{ handlungsabsicht initialwelt (Gesetz \{\})} \rangle$

14.3 Alice erzeugt 5 Wohlstand für sich.

Wir definieren eine Maxime die besagt, dass sich der Besitz einer Person nicht verringern darf:

fun $\text{individueller-fortschritt} :: \langle \text{person} \Rightarrow \text{zahlenwelt handlung} \Rightarrow \text{bool} \rangle$ **where**
 $\langle \text{individueller-fortschritt } p (\text{Handlung vor nach}) \longleftrightarrow (\text{meins } p \text{ vor}) \leq (\text{meins } p \text{ nach}) \rangle$
definition $\text{maxime-zahlenfortschritt} :: \langle (\text{person}, \text{zahlenwelt}) \text{ maxime} \rangle$ **where**
 $\langle \text{maxime-zahlenfortschritt} \equiv \text{Maxime } (\lambda \text{ich. individueller-fortschritt ich}) \rangle$

lemma $\langle \text{maxime-und-handlungsabsicht-generalisieren maxime-zahlenfortschritt (HandlungF (erschaffen 5))}$
 $p \rangle$

lemma $\langle \text{maxime-und-handlungsabsicht-generalisieren maxime-zahlenfortschritt (HandlungF (stehlen 5 Bob))}$
 $p \rangle$

lemma $\text{mhg-maxime-zahlenfortschritt-stehlen4}:$
 $\langle \text{maxime-und-handlungsabsicht-generalisieren maxime-zahlenfortschritt (HandlungF (stehlen4 1 10)) } p \rangle$

In jeder Welt ist die Handlung *moralisch*:

lemma $\langle \text{moralisch welt maxime-zahlenfortschritt (HandlungF (erschaffen 5))} \rangle$

Die *maxime-zahlenfortschritt* erfüllt nicht den *kategorischer-imperativ* da *Alice* nach der Maxime z.B. *Bob* bestehlen würde.

lemma $\langle \neg \text{kategorischer-imperativ zahlenwelt-personen-swap initialwelt maxime-zahlenfortschritt} \rangle$

lemma $\text{hlp1: } \langle \text{meins } p1 (\text{zahlenwelt-personen-swap } p1 p2 \text{ welt}) = \text{meins } p2 \text{ welt} \rangle$

lemma $\text{hlp2: } \langle \text{meins } p2 (\text{zahlenwelt-personen-swap } p1 p2 \text{ welt}) = \text{meins } p1 \text{ welt} \rangle$

lemma $\text{hlp3: } \langle p1 \neq p2 \implies p \neq p1 \implies p \neq p2 \implies$
 $\text{meins } p (\text{zahlenwelt-personen-swap } p1 p2 \text{ welt}) = \text{meins } p \text{ welt} \rangle$

lemma $\langle \text{wpsm-kommutiert (Maxime individueller-fortschritt) zahlenwelt-personen-swap welt} \rangle$

lemma $\langle \text{wpsm-kommutiert}$
 $(\text{Maxime } (\lambda (\text{ich}::\text{person}) h. (\forall pX. \text{individueller-fortschritt } pX h)))$
 $\text{zahlenwelt-personen-swap welt} \rangle$

lemma $\langle \textit{kategorischer-imperativ zahlenwelt-personen-swap welt} \\ (\textit{Maxime } (\lambda(\textit{ich}::\textit{person}) h. (\forall pX. \textit{individueller-fortschritt } pX h))) \rangle$

Alice kann beliebig oft 5 Wohlstand für sich selbst erschaffen. Das entstehende Gesetz ist nicht sehr gut, da es einfach jedes Mal einen Snapshot der Welt aufschreibt und nicht sehr generisch ist.

lemma $\langle \textit{beispiel-case-law-absolut maxime-zahlenfortschritt (HandlungF (erschaffen 5))} \\ = \\ \textit{Gesetz} \\ \{(\S 5, \\ \textit{Rechtsnorm} \\ (\textit{Tatbestand } ([(\textit{Alice}, 25), (\textit{Bob}, 10), (\textit{Carol}, - 3)], [(\textit{Alice}, 30), (\textit{Bob}, 10), (\textit{Carol}, - 3)])) \\ (\textit{Rechtsfolge Erlaubnis})), \\ (\S 4, \\ \textit{Rechtsnorm} \\ (\textit{Tatbestand } ([(\textit{Alice}, 20), (\textit{Bob}, 10), (\textit{Carol}, - 3)], [(\textit{Alice}, 25), (\textit{Bob}, 10), (\textit{Carol}, - 3)])) \\ (\textit{Rechtsfolge Erlaubnis})), \\ (\S 3, \\ \textit{Rechtsnorm} \\ (\textit{Tatbestand } ([(\textit{Alice}, 15), (\textit{Bob}, 10), (\textit{Carol}, - 3)], [(\textit{Alice}, 20), (\textit{Bob}, 10), (\textit{Carol}, - 3)])) \\ (\textit{Rechtsfolge Erlaubnis})), \\ (\S 2, \\ \textit{Rechtsnorm} \\ (\textit{Tatbestand } ([(\textit{Alice}, 10), (\textit{Bob}, 10), (\textit{Carol}, - 3)], [(\textit{Alice}, 15), (\textit{Bob}, 10), (\textit{Carol}, - 3)])) \\ (\textit{Rechtsfolge Erlaubnis})), \\ (\S 1, \\ \textit{Rechtsnorm} \\ (\textit{Tatbestand } ([(\textit{Alice}, 5), (\textit{Bob}, 10), (\textit{Carol}, - 3)], [(\textit{Alice}, 10), (\textit{Bob}, 10), (\textit{Carol}, - 3)])) \\ (\textit{Rechtsfolge Erlaubnis}))) \} \\ \rangle$

Die gleiche Handlung, wir schreiben aber nur die Änderung der Welt ins Gesetz:

lemma $\langle \textit{beispiel-case-law-relativ maxime-zahlenfortschritt (HandlungF (erschaffen 5))} = \\ \textit{Gesetz} \\ \{(\S 1, \textit{Rechtsnorm } (\textit{Tatbestand } [\textit{Gewinnt Alice 5}] (\textit{Rechtsfolge Erlaubnis}))) \} \rangle$

lemma $\langle \textit{beispiel-case-law-relativ} \\ (\textit{Maxime } (\lambda(\textit{ich}::\textit{person}) h. (\forall pX. \textit{individueller-fortschritt } pX h))) (\textit{HandlungF (erschaffen 5)}) = \\ \textit{Gesetz } \{(\S 1, \textit{Rechtsnorm } (\textit{Tatbestand } [\textit{Gewinnt Alice 5}] (\textit{Rechtsfolge Erlaubnis}))) \} \rangle$

14.4 Kleine Änderung in der Maxime

In der Maxime *individueller-fortschritt* hatten wir *meins p vor* \leq *meins p nach*. Was wenn wir nun echten Fortschritt fordern: *meins p vor* $<$ *meins p nach*.

fun *individueller-strikter-fortschritt* :: $\langle \text{person} \Rightarrow \text{zahlenwelt handlung} \Rightarrow \text{bool} \rangle$ **where**
 $\langle \text{individueller-strikter-fortschritt } p \text{ (Handlung vor nach)} \longleftrightarrow (\text{meins } p \text{ vor}) < (\text{meins } p \text{ nach}) \rangle$

Nun ist es *Alice* verboten Wohlstand für sich selbst zu erzeugen.

lemma $\langle \text{beispiel-case-law-relativ}$
 $(\text{Maxime } (\lambda \text{ich. individueller-strikter-fortschritt ich}))$
 $(\text{HandlungF (erschaffen 5)}) =$
 $\text{Gesetz } \{(\$ 1, \text{Rechtsnorm (Tatbestand [Gewinnt Alice 5]) (Rechtsfolge Verbot)})\} \rangle$

In keiner Welt ist die Handlung nun *moralisch*:

lemma $\langle \neg \text{moralisch welt}$
 $(\text{Maxime } (\lambda \text{ich. individueller-strikter-fortschritt ich})) (\text{HandlungF (erschaffen 5)}) \rangle$

Der Grund ist, dass der Rest der Bevölkerung keine *strikte* Erhöhung des eigenen Wohlstands erlebt. Effektiv führt diese Maxime zu einem Gesetz, welches es einem Individuum nicht erlaubt mehr Besitz zu erschaffen, obwohl niemand dadurch einen Nachteil hat. Diese Maxime kann meiner Meinung nach nicht gewollt sein.

Beispielsweise ist *Bob* das Opfer wenn *Alice* sich 5 Wohlstand erschafft, aber *Bob's* Wohlstand sich nicht erhöht:

lemma $\langle \text{VerletzteMaxime (Opfer Bob) (Taeter Alice)}$
 $(\text{Handlung } [(Alice, 5), (Bob, 10), (Carol, -3)] [(Alice, 10), (Bob, 10), (Carol, -3)])$
 $\in \text{debug-maxime show-zahlenwelt initialwelt}$
 $(\text{Maxime } (\lambda \text{ich. individueller-strikter-fortschritt ich})) (\text{HandlungF (erschaffen 5)}) \rangle$

14.5 Maxime für Globales Optimum

Wir bauen nun eine Maxime, die das Individuum vernachlässigt und nur nach dem globalen Optimum strebt:

fun *globaler-strikter-fortschritt* :: $\langle \text{zahlenwelt handlung} \Rightarrow \text{bool} \rangle$ **where**
 $\langle \text{globaler-strikter-fortschritt (Handlung vor nach)} \longleftrightarrow (\text{gesamtbesitz vor}) < (\text{gesamtbesitz nach}) \rangle$

Die Maxime ignoriert das *ich* komplett.

Nun ist es *Alice* wieder erlaubt, Wohlstand für sich selbst zu erzeugen, da sich dadurch auch der Gesamtwohlstand erhöht:

lemma $\langle \text{beispiel-case-law-relativ}$
 $(\text{Maxime } (\lambda \text{ich. globaler-strikter-fortschritt}))$
 $(\text{HandlungF (erschaffen 5)}) =$
 $\text{Gesetz } \{(\$ 1, \text{Rechtsnorm (Tatbestand [Gewinnt Alice 5]) (Rechtsfolge Erlaubnis)})\} \rangle$

lemma $\langle \text{moralisch initialwelt}$
 $(\text{Maxime } (\lambda \text{ich. globaler-strikter-fortschritt})) (\text{HandlungF (erschaffen 5)}) \rangle$

Allerdings ist auch diese Maxime auch sehr grausam, da sie Untätigkeit verbietet:

```
lemma <beispiel-case-law-relativ
  (Maxime ( $\lambda$ ich. globaler-strikter-fortschritt))
  (HandlungF (erschaffen 0)) =
  Gesetz {(\$ 1, Rechtsnorm (Tatbestand []) (Rechtsfolge Verbot))}>
```

Unsere initiale einfache *maxime-zahlenfortschritt* würde Untätigkeit hier erlauben:

```
lemma <beispiel-case-law-relativ
  maxime-zahlenfortschritt
  (HandlungF (erschaffen 0)) =
  Gesetz {(\$ 1, Rechtsnorm (Tatbestand []) (Rechtsfolge Erlaubnis))}>
```

Wir können die Maxime für globalen Fortschritt etwas lockern:

```
fun globaler-fortschritt :: <zahlenwelt handlung  $\Rightarrow$  bool> where
  <globaler-fortschritt (Handlung vor nach)  $\longleftrightarrow$  (gesamtbesitz vor)  $\leq$  (gesamtbesitz nach)>
```

Untätigkeit ist nun auch hier erlaubt:

```
lemma <beispiel-case-law-relativ
  (Maxime ( $\lambda$ ich. globaler-fortschritt))
  (HandlungF (erschaffen 0))
  =
  Gesetz {(\$ 1, Rechtsnorm (Tatbestand []) (Rechtsfolge Erlaubnis))}>
```

```
lemma < $\neg$ wohlgeformte-handlungsabsicht
  zahlenwelt-personen-swap initialwelt
  (HandlungF ( $\lambda$ ich w. if ich = Alice then w else Zahlenwelt ( $\lambda$ -. 0)))>
```

```
lemma < $\neg$  maxime-und-handlungsabsicht-generalisieren maxime-zahlenfortschritt
  (HandlungF ( $\lambda$ ich w. if ich = Alice then w else Zahlenwelt ( $\lambda$ -. 0))) Carol>
theorem <kategorischer-imperativ zahlenwelt-personen-swap (Zahlenwelt besitz)
  (Maxime ( $\lambda$ ich::person. globaler-fortschritt))>
```

Allerdings ist auch Stehlen erlaubt, da global gesehen, kein Besitz vernichtet wird:

```
lemma <beispiel-case-law-relativ
  (Maxime ( $\lambda$ ich. globaler-fortschritt))
  (HandlungF (stehlen 5 Bob))
  =
  Gesetz
  {(\$ 1, Rechtsnorm (Tatbestand [Gewinnt Alice 5, Verliert Bob 5]) (Rechtsfolge Erlaubnis))}>
```

14.6 Alice stiehlt 5

Zurück zur einfachen *maxime-zahlenfortschritt*.

Stehlen ist verboten:

lemma $\langle \text{beispiel-case-law-relativ maxime-zahlenfortschritt (HandlungF (stehlen 5 Bob))} =$
Gesetz
 $\{(\S 1, \text{Rechtsnorm (Tatbestand [Gewinnt Alice 5, Verliert Bob 5]) (Rechtsfolge Verbot)})\} \rangle$

In kein Welt ist Stehlen *moralisch*:

lemma $\langle \neg \text{moralisch welt maxime-zahlenfortschritt (HandlungF (stehlen 5 Bob))} \rangle$

Auch wenn *Alice* von sich selbst stehlen möchte ist dies verboten, obwohl hier keiner etwas verliert:

lemma $\langle \text{beispiel-case-law-relativ maxime-zahlenfortschritt (HandlungF (stehlen 5 Alice))} =$
Gesetz $\{(\S 1, \text{Rechtsnorm (Tatbestand []) (Rechtsfolge Verbot)})\} \rangle$

Der Grund ist, dass *Alice* die abstrakte Handlung "Alice wird bestohlen" gar nicht gut fände, wenn sie jemand anderes ausführt:

lemma $\langle \text{debug-maxime show-zahlenwelt initialwelt}$
 $\text{maxime-zahlenfortschritt (HandlungF (stehlen 5 Alice))} =$
 $\{ \text{VerletzteMaxime (Opfer Alice) (Taeter Bob)}$
 $(\text{Handlung [(Alice, 5), (Bob, 10), (Carol, - 3)] [(Bob, 15), (Carol, - 3)]),$
 $\text{VerletzteMaxime (Opfer Alice) (Taeter Carol)}$
 $(\text{Handlung [(Alice, 5), (Bob, 10), (Carol, - 3)] [(Bob, 10), (Carol, 2)]),$
 $\text{VerletzteMaxime (Opfer Alice) (Taeter Eve)}$
 $(\text{Handlung [(Alice, 5), (Bob, 10), (Carol, - 3)] [(Bob, 10), (Carol, - 3), (Eve, 5)])}$
 $\} \rangle$

Leider ist das hier abgeleitete Gesetz sehr fragwürdig: *Rechtsnorm (Tatbestand []) (Rechtsfolge Verbot)*

Es besagt, dass Nichtstun verboten ist.

Indem wir die beiden Handlungen Nichtstun und Selbstbestehlen betrachten, können wir sogar ein widersprüchliches Gesetz ableiten:

lemma $\langle \text{simulateOne}$
 $(\text{SimConsts}$
 Alice
 $\text{maxime-zahlenfortschritt}$
 $(\text{case-law-ableiten-relativ delta-zahlenwelt}))$
 $20 (\text{HandlungF (stehlen 5 Alice)}) \text{ initialwelt}$
 $(\text{beispiel-case-law-relativ maxime-zahlenfortschritt (HandlungF (erschaffen 0))})$
 $=$
Gesetz
 $\{(\S 2, \text{Rechtsnorm (Tatbestand []) (Rechtsfolge Verbot)}),$
 $(\S 1, \text{Rechtsnorm (Tatbestand []) (Rechtsfolge Erlaubnis)})\} \rangle$

Meine persönliche Conclusion: Wir müssen irgendwie die Absicht mit ins Gesetz schreiben.

14.7 Schenken

Es ist *Alice* verboten, etwas zu verschenken:

lemma $\langle \text{beispiel-case-law-relativ maxime-zahlenfortschritt } (HandlungF \text{ (schenken 5 Bob)})$
 $=$
Gesetz
 $\{(\S \text{ 1,}$
 $\text{Rechtsnorm (Tatbestand [Verliert Alice 5, Gewinnt Bob 5]) (Rechtsfolge Verbot)})\}$

Der Grund ist, dass *Alice* dabei etwas verliert und die *maxime-zahlenfortschritt* dies nicht Erlaubt. Es fehlt eine Möglichkeit zu modellieren, dass *Alice* damit einverstanden ist, etwas abzugeben. Doch wir haben bereits in *stehlen* $i = \text{schenken } (- i)$ gesehen, dass *stehlen* und *schenken* nicht unterscheidbar sind.

14.8 Ungültige Maxime

Es ist verboten, in einer Maxime eine spezielle Person hardzucoden. Da dies gegen die Gleichbehandlung aller Menschen verstoßen würde.

Beispielsweise könnten wir *individueller-fortschritt* nicht mehr parametrisiert verwenden, sondern einfach *Alice* reinschreiben:

lemma $\langle \text{individueller-fortschritt Alice}$
 $= (\lambda h. \text{ case } h \text{ of } Handlung \text{ vor nach} \Rightarrow (\text{meins Alice vor}) \leq (\text{meins Alice nach})) \rangle$

Dies würde es erlauben, dass *Alice* Leute bestehlen darf:

lemma $\langle \text{beispiel-case-law-relativ}$
 $(\text{Maxime } (\lambda ich. \text{ individueller-fortschritt Alice}))$
 $(\text{HandlungF (stehlen 5 Bob)})$
 $=$
Gesetz
 $\{(\S \text{ 1, Rechtsnorm (Tatbestand [Gewinnt Alice 5, Verliert Bob 5]) (Rechtsfolge Erlaubnis)})\}$

15 Einkommensteuergesetzgebung

Basierend auf einer stark vereinfachten Version des deutschen Steuerrechts. Wenn ich Wikipedia richtig verstanden habe, habe ich sogar aus Versehen einen Teil des österreichischen Steuersystem gebaut mit deutschen Konstanten.

Folgende **locale** nimmt an, dass wir eine Funktion $\text{steuer}::\text{nat} \Rightarrow \text{nat}$ haben, welche basierend auf dem Einkommen die zu zahlende Steuer berechnet.

Die *steuer* Funktion arbeitet auf natürlichen Zahlen. Wir nehmen an, dass einfach immer auf ganze Geldbeträge gerundet wird. Wie im deutschen System.

Die **locale** enthält einige Definition, gegeben die *steuer* Funktion.

Eine konkrete *steuer* Funktion wird noch nicht gegeben.

```

locale steuer-defs =
  fixes steuer ::  $\langle \text{nat} \Rightarrow \text{nat} \rangle$  — Einkommen  $\rightarrow$  Steuer
begin
  definition brutto ::  $\langle \text{nat} \Rightarrow \text{nat} \rangle$  where
     $\langle \text{brutto einkommen} \equiv \text{einkommen} \rangle$ 
  definition netto ::  $\langle \text{nat} \Rightarrow \text{nat} \rangle$  where
     $\langle \text{netto einkommen} \equiv \text{einkommen} - (\text{steuer einkommen}) \rangle$ 
  definition steuersatz ::  $\langle \text{nat} \Rightarrow \text{percentage} \rangle$  where
     $\langle \text{steuersatz einkommen} \equiv \text{percentage } ((\text{steuer einkommen}) / \text{einkommen}) \rangle$ 
end

```

Beispiel. Die *steuer* Funktion sagt, man muss 25 Prozent Steuern zahlen:

```

definition beispiel-25prozent-steuer ::  $\langle \text{nat} \Rightarrow \text{nat} \rangle$  where
   $\langle \text{beispiel-25prozent-steuer } e \equiv \text{nat } \lfloor \text{real } e * (\text{percentage } 0.25) \rfloor \rangle$ 

```

```

lemma
   $\langle \text{beispiel-25prozent-steuer } 100 = 25 \rangle$ 
   $\langle \text{steuer-defs.brutto } 100 = 100 \rangle$ 
   $\langle \text{steuer-defs.netto beispiel-25prozent-steuer } 100 = 75 \rangle$ 
   $\langle \text{steuer-defs.steuersatz beispiel-25prozent-steuer } 100 = \text{percentage } 0.25 \rangle$ 

```

Folgende **locale** erweitert die *steuer-defs locale* und stellt einige Anforderungen die eine gültige *steuer* Funktion erfüllen muss.

- Wer mehr Einkommen hat, muss auch mehr Steuern zahlen.
- Leistung muss sich lohnen: Wer mehr Einkommen hat muss auch nach Abzug der Steuer mehr übrig haben.
- Existenzminimum: Es gibt ein Existenzminimum, welches nicht besteuert werden darf.

```

locale steuersystem = steuer-defs +
  assumes wer-hat-der-gibt:
     $\langle \text{einkommen-a} \geq \text{einkommen-b} \implies \text{steuer einkommen-a} \geq \text{steuer einkommen-b} \rangle$ 

```

```

and leistung-lohnt-sich:
   $\langle \text{einkommen-a} \geq \text{einkommen-b} \implies \text{netto einkommen-a} \geq \text{netto einkommen-b} \rangle$ 

```

— Ein Existenzminimum wird nicht versteuert. Zahl Deutschland 2022, vermutlich sogar die falsche Zahl.

```

and existenzminimum:
   $\langle \text{einkommen} \leq 9888 \implies \text{steuer einkommen} = 0 \rangle$ 

```

```

begin

```

```

end

```

Eigentlich hätte ich gerne noch eine weitere Anforderung. <https://de.wikipedia.org/wiki/Steuerprogression> sagt "Steuerprogression bedeutet das Ansteigen des Steuersatzes in Abhängigkeit vom zu versteuernden Einkommen oder Vermögen."

Formal betrachtet würde das bedeuten $\text{einkommen-}b \leq \text{einkommen-}a \implies (\lambda x. \text{real-of-percentage } (\text{steuer-defs.steuersatz } \text{einkommen-}b \ x)) \leq (\lambda x. \text{real-of-percentage } (\text{steuer-defs.steuersatz } \text{einkommen-}a \ x))$

Leider haben wir bereits jetzt in dem Modell eine Annahme getroffen, die es uns quasi unmöglich macht, ein Steuersystem zu implementieren, welches die Steuerprogression erfüllt. Der Grund ist, dass wir die Steuerfunktion auf ganzen Zahlen definiert haben. Aufgrund von Rundung können wir also immer Fälle haben, indem ein höheres Einkommen einen leicht geringeren Steuersatz hat als ein geringeres Einkommen. Beispielsweise bedeutet das für *beispiel-25prozent-steuer*, dass jemand mit 100 EUR Einkommen genau 25 Prozent Steuer zahlt, jemand mit 103 EUR Einkommen aber nur ca 24,3 Prozent Steuer zahlt.

lemma

```
⟨steuer-defs.steuersatz beispiel-25prozent-steuer 100 = percentage 0.25⟩
⟨steuer-defs.steuersatz beispiel-25prozent-steuer 103 = percentage (25 / 103)⟩
⟨percentage (25 / 103) < percentage 0.25⟩
⟨(103::nat) > 100⟩
```

In der Praxis sollten diese kleinen Rundungsfehler kein Problem darstellen, in diesem theoretischen Modell sorgen sie aber dafür, dass unser Steuersystem (und wir modellieren eine vereinfachte Version des deutschen Steuersystems) keine Steuerprogression erfüllt.

Die folgende Liste, basierend auf [https://de.wikipedia.org/wiki/Einkommensteuer_\(Deutschland\)#Tarif_2022](https://de.wikipedia.org/wiki/Einkommensteuer_(Deutschland)#Tarif_2022), sagt in welchem Bereich welcher Prozentsatz an Steuern zu zahlen ist. Beispielsweise sind die ersten 10347 steuerfrei.

definition *steuerbuckets2022* :: $\langle (\text{nat} \times \text{percentage}) \text{ list} \rangle$ **where**

```
⟨steuerbuckets2022 ≡ [
  (10347, percentage 0),
  (14926, percentage 0.14),
  (58596, percentage 0.2397),
  (277825, percentage 0.42)
]⟩
```

Für jedes Einkommen über 277825 gilt der Spitzensteuersatz von 45 Prozent. Wir ignorieren die Progressionsfaktoren in Zone 2 und 3.

Folgende Funktion berechnet die zu zahlende Steuer, basierend auf einer Steuerbucketliste.

fun *bucketsteuerAbs* :: $\langle (\text{nat} \times \text{percentage}) \text{ list} \Rightarrow \text{percentage} \Rightarrow \text{nat} \Rightarrow \text{real} \rangle$ **where**

```
⟨bucketsteuerAbs ((bis, prozent)#mehr) spitzensteuer e =
  ((min bis e) * prozent)
  + (bucketsteuerAbs (map (λ(s,p). (s-bis,p)) mehr) spitzensteuer (e - bis))⟩
| ⟨bucketsteuerAbs [] spitzensteuer e = e*spitzensteuer⟩
```

Die Einkommenssteuerberechnung, mit Spitzensteuersatz 45 Prozent und finalem Abrunden.

definition *einkommenssteuer* :: $\langle \text{nat} \Rightarrow \text{nat} \rangle$ **where**
 $\langle \text{einkommenssteuer } \text{einkommen} \equiv$
 $\text{floor } (\text{bucketsteuerAbs } \text{steuerbuckets2022 } (\text{percentage } 0.45) \text{ einkommen}) \rangle$

Beispiel. Alles unter dem Existenzminimum ist steuerfrei:

lemma $\langle \text{einkommenssteuer } 10 = 0 \rangle$

lemma $\langle \text{einkommenssteuer } 10000 = 0 \rangle$

Für ein Einkommen nur knapp über dem Existenzminimum fällt sehr wenig Steuer an:

lemma $\langle \text{einkommenssteuer } 14000 = \text{floor } ((14000 - 10347) * 0.14) \rangle$

lemma $\langle \text{einkommenssteuer } 14000 = 511 \rangle$

Bei einem Einkommen von 20000 EUR wird ein Teil bereits mit den höheren Steuersatz der 3. Zone besteuert:

lemma $\langle \text{einkommenssteuer } 20000 = 1857 \rangle$

lemma $\langle \text{einkommenssteuer } 20000 =$
 $\text{floor } ((14926 - 10347) * 0.14 + (20000 - 14926) * 0.2397) \rangle$

Höhere Einkommen führen zu einer höheren Steuer:

lemma $\langle \text{einkommenssteuer } 40000 = 6651 \rangle$

lemma $\langle \text{einkommenssteuer } 60000 = 11698 \rangle$

Die *einkommenssteuer* Funktion erfüllt die Anforderungen an *steuersystem*.

interpretation *steuersystem*
where *steuer* = $\langle \text{einkommenssteuer} \rangle$

16 Beispiel: Steuern

Wir nehmen eine einfach Welt an, in der jeder Person ihr Einkommen zugeordnet wird.

Achtung: Im Unterschied zum BeispielZahlenwelt.thy modellieren wir hier nicht den Gesamtbesitz, sondern das Jahreseinkommen. Besitz wird ignoriert.

datatype *steuerwelt* = *Steuerwelt*
 $(\text{get-einkommen}: \langle \text{person} \Rightarrow \text{int} \rangle)$ — einkommen jeder Person (im Zweifel 0).

fun *steuerlast* :: $\langle \text{person} \Rightarrow \text{steuerwelt } \text{handlung} \Rightarrow \text{int} \rangle$ **where**
 $\langle \text{steuerlast } p (\text{Handlung vor nach}) = ((\text{get-einkommen vor}) p) - ((\text{get-einkommen nach}) p) \rangle$

fun *brutto* :: $\langle \text{person} \Rightarrow \text{steuerwelt } \text{handlung} \Rightarrow \text{int} \rangle$ **where**
 $\langle \text{brutto } p (\text{Handlung vor nach}) = (\text{get-einkommen vor}) p \rangle$

fun *netto* :: $\langle \text{person} \Rightarrow \text{steuerwelt } \text{handlung} \Rightarrow \text{int} \rangle$ **where**
 $\langle \text{netto } p (\text{Handlung vor nach}) = (\text{get-einkommen nach}) p \rangle$

lemma $\langle \text{steuerlast Alice } (\text{Handlung } (\text{Steuerwelt } \clubsuit[\text{Alice}:=8]) (\text{Steuerwelt } \clubsuit[\text{Alice}:=5])) = 3 \rangle$

lemma $\langle \text{steuerlast Alice } (\text{Handlung } (\text{Steuerwelt } \clubsuit[\text{Alice}:=8]) (\text{Steuerwelt } \clubsuit[\text{Alice}:=0])) = 8 \rangle$

```

lemma <steuerlast Bob (Handlung (Steuerwelt ♣[Alice:=8]) (Steuerwelt ♣[Alice:=5])) = 0>
lemma <steuerlast Alice (Handlung (Steuerwelt ♣[Alice:=-3]) (Steuerwelt ♣[Alice:=-4])) = 1>
lemma <steuerlast Alice (Handlung (Steuerwelt ♣[Alice:=1]) (Steuerwelt ♣[Alice:=-1])) = 2>

fun mehrverdiener :: <person ⇒ steuerwelt handlung ⇒ person set> where
  <mehrverdiener ich (Handlung vor nach) = {p. (get-einkommen vor) p ≥ (get-einkommen vor) ich}>

lemma <mehrverdiener Alice
  (Handlung (Steuerwelt ♣[Alice:=8, Bob:=12, Eve:=7]) (Steuerwelt ♣[Alice:=5]))
  = {Alice, Bob}>

```

Folgende Maxime versucht Steuergerechtigkeit festzuschreiben:

```

definition maxime-steuern :: <(person, steuerwelt) maxime> where
  <maxime-steuern ≡ Maxime
    (λich handlung.
      (∀ p∈mehrverdiener ich handlung.
        steuerlast ich handlung ≤ steuerlast p handlung)
      ∧ (∀ p∈mehrverdiener ich handlung.
        netto ich handlung ≤ netto p handlung)
    )>

```

16.1 Setup für Beispiele

```

definition <initialwelt ≡ Steuerwelt ♣[Alice:=8, Bob:=3, Eve:= 5]>

```

```

definition <beispiel-case-law-absolut welt steuerfun ≡
  simulateOne
    (SimConsts
      Alice
      maxime-steuern
      (printable-case-law-ableiten-absolut (λw. show-fun (get-einkommen w))))
  3 steuerfun welt (Gesetz {})>

```

```

definition <beispiel-case-law-relativ welt steuerfun ≡
  simulateOne
    (SimConsts
      Alice
      maxime-steuern
      (case-law-ableiten-relativ delta-steuerwelt))
  1 steuerfun welt (Gesetz {})>

```

16.2 Beispiel: Keiner Zahlt Steuern

Die Maxime ist erfüllt, da wir immer nur kleiner-gleich fordern!

```

lemma <beispiel-case-law-relativ initialwelt (HandlungF (λich welt. welt)) =
  Gesetz { (§ 1, Rechtsnorm (Tatbestand []) (Rechtsfolge Erlaubnis))}>

```

16.3 Beispiel: Ich zahle 1 Steuer

Das funktioniert nicht:

definition $\langle \text{ich-zahle-1-steuer } \text{ich } \text{welt} \equiv$

$\text{Steuerwelt } ((\text{get-einkommen } \text{welt})(\text{ich} \text{ } \text{--} \text{ } 1)) \rangle$

lemma $\langle \text{beispiel-case-law-absolut } \text{initialwelt } (\text{HandlungF } \text{ich-zahle-1-steuer}) =$

Gesetz

$\{(\S \text{ } 1,$

Rechtsnorm

$(\text{Tatbestand}$

$[(\text{Alice}, 8), (\text{Bob}, 3), (\text{Carol}, 0), (\text{Eve}, 5)],$

$[(\text{Alice}, 7), (\text{Bob}, 3), (\text{Carol}, 0), (\text{Eve}, 5)])$

$(\text{Rechtsfolge } \text{Verbot})) \}$

lemma $\langle \text{beispiel-case-law-relativ } \text{initialwelt } (\text{HandlungF } \text{ich-zahle-1-steuer}) =$

Gesetz

$\{(\S \text{ } 1, \text{Rechtsnorm } (\text{Tatbestand } [\text{Verliert Alice } 1])$

$(\text{Rechtsfolge } \text{Verbot})) \}$

Denn jeder muss Steuer zahlen! Ich finde es super spannend, dass hier faktisch ein Gleichbehandlungsgrundsatz rausfällt, ohne dass wir soewtas jemals explizit gefordert haben.

16.4 Beispiel: Jeder zahle 1 Steuer

Jeder muss steuern zahlen: funktioniert, ist aber doof, denn am Ende sind alle im Minus.

Das *ich* wird garnicht verwendet, da jeder Steuern zahlt.

definition $\langle \text{jeder-zahle-1-steuer } \text{ich } \text{welt} \equiv$

$\text{Steuerwelt } ((\lambda e. e - 1) \circ (\text{get-einkommen } \text{welt})) \rangle$

lemma $\langle \text{beispiel-case-law-absolut } \text{initialwelt } (\text{HandlungF } \text{jeder-zahle-1-steuer}) =$

Gesetz

$\{(\S \text{ } 3,$

Rechtsnorm

$(\text{Tatbestand}$

$[(\text{Alice}, 6), (\text{Bob}, 1), (\text{Carol}, -2), (\text{Eve}, 3)],$

$[(\text{Alice}, 5), (\text{Bob}, 0), (\text{Carol}, -3), (\text{Eve}, 2)])$

$(\text{Rechtsfolge } \text{Erlaubnis})),$

$(\S \text{ } 2,$

Rechtsnorm

$(\text{Tatbestand}$

$[(\text{Alice}, 7), (\text{Bob}, 2), (\text{Carol}, -1), (\text{Eve}, 4)],$

$[(\text{Alice}, 6), (\text{Bob}, 1), (\text{Carol}, -2), (\text{Eve}, 3)])$

$(\text{Rechtsfolge } \text{Erlaubnis})),$

$(\S \text{ } 1,$

Rechtsnorm

$(\text{Tatbestand}$

$[(\text{Alice}, 8), (\text{Bob}, 3), (\text{Carol}, 0), (\text{Eve}, 5)],$

$[(\text{Alice}, 7), (\text{Bob}, 2), (\text{Carol}, -1), (\text{Eve}, 4)])$

$(\text{Rechtsfolge } \text{Erlaubnis})) \}$

lemma $\langle \text{beispiel-case-law-relativ } \text{initialwelt } (\text{HandlungF } \text{jeder-zahle-1-steuer}) =$

Gesetz
 $\{(\S\ 1,$
Rechtsnorm
 $(\text{Tatbestand } [\text{Verliert Alice } 1, \text{ Verliert Bob } 1, \text{ Verliert Carol } 1, \text{ Verliert Eve } 1])$
 $(\text{Rechtsfolge Erlaubnis}))\}$

16.5 Beispiel: Vereinfachtes Deutsches Steuersystem

Jetzt kommt die *Steuern.thy* ins Spiel.

definition *jeder-zahlt* :: $\langle (\text{nat} \Rightarrow \text{nat}) \Rightarrow 'a \Rightarrow \text{steuerwelt} \Rightarrow \text{steuerwelt} \rangle$ **where**
 $\langle \text{jeder-zahlt steuerberechnung ich welt} \equiv$
 $\text{Steuerwelt } ((\lambda e. e - \text{steuerberechnung } e) \circ \text{nat} \circ (\text{get-einkommen } \text{welt})) \rangle$

definition $\langle \text{jeder-zahlt-einkommenssteuer} \equiv \text{jeder-zahlt einkommenssteuer} \rangle$

Bei dem geringen Einkommen der *initialwelt* zahlt keiner Steuern.

lemma $\langle \text{beispiel-case-law-absolut initialwelt } (\text{HandlungF } \text{jeder-zahlt-einkommenssteuer}) =$
Gesetz
 $\{(\S\ 1,$
Rechtsnorm
 $(\text{Tatbestand}$
 $[(\text{Alice}, 8), (\text{Bob}, 3), (\text{Carol}, 0), (\text{Eve}, 5)],$
 $[(\text{Alice}, 8), (\text{Bob}, 3), (\text{Carol}, 0), (\text{Eve}, 5)])$
 $(\text{Rechtsfolge Erlaubnis}))\}$

Für höhere Einkommen erhalten wir plausible Werte und niemand rutscht ins negative:

lemma $\langle \text{beispiel-case-law-relativ}$
 $(\text{Steuerwelt } \clubsuit [\text{Alice}:=10000, \text{Bob}:=14000, \text{Eve}:=20000])$
 $(\text{HandlungF } \text{jeder-zahlt-einkommenssteuer})$
 $=$
Gesetz
 $\{(\S\ 1,$
Rechtsnorm $(\text{Tatbestand } [\text{Verliert Bob } 511, \text{ Verliert Eve } 1857])$
 $(\text{Rechtsfolge Erlaubnis}))\}$

17 Vereinfachtes Deutsches Steuersystem vs. die Steuermaxime

Die Anforderungen für ein *steuersystem* und die *maxime-steuern* sind vereinbar.

lemma *steuersystem-imp-maxime*:
 $\langle \text{steuersystem steuersystem-impl} \implies$
 $(\forall \text{ welt. moralisch welt maxime-steuern } (\text{HandlungF } (\text{jeder-zahlt steuersystem-impl}))) \rangle$

Danke ihr nats. Macht also keinen Sinn das als Annahme in die Maxime zu packen....

lemma *steuern-kleiner-einkommen-nat*:
 $\langle \text{steuerlast ich } (\text{Handlung welt } (\text{jeder-zahlt steuersystem-impl ich welt}))$
 $\leq \text{brutto ich } (\text{Handlung welt } (\text{jeder-zahlt steuersystem-impl ich welt})) \rangle$

lemma *maxime-imp-steuersystem*:

$\langle (\forall \text{einkommen. steuersystem-impl einkommen} \leq \text{einkommen}) \implies$
 $(\forall \text{einkommen. einkommen} \leq 9888 \longrightarrow \text{steuersystem-impl einkommen} = 0) \implies$
 $\forall \text{welt. moralisch welt maxime-steuern (HandlungF (jeder-zahlt steuersystem-impl))}$
 $\implies \text{steuersystem steuersystem-impl} \rangle$

Für jedes *steuersystem-impl::nat* \Rightarrow *nat*, mit zwei weiteren Annahmen, gilt das *steuersystem* und *maxime-steuern* in der *jeder-zahlt* Implementierung äquivalent sind.

theorem

fixes *steuersystem-impl* :: $\langle \text{nat} \Rightarrow \text{nat} \rangle$

assumes *steuer-kleiner-einkommen*: $\langle \forall \text{einkommen. steuersystem-impl einkommen} \leq \text{einkommen} \rangle$

and *existenzminimum*: $\langle \forall \text{einkommen. einkommen} \leq 9888 \longrightarrow \text{steuersystem-impl einkommen} = 0 \rangle$

shows

$\langle (\forall \text{welt. moralisch welt maxime-steuern (HandlungF (jeder-zahlt steuersystem-impl)))}$
 $\longleftrightarrow \text{steuersystem steuersystem-impl} \rangle$